
Goal setting 

 The Excel schedule should have the following skills: 

1. Automatic update of the group table 

After each entry of a match result the group table is automatically updated. 

All UEFA rules are taken into account, including fair play and European Qualifiers Rankings. 

The teams qualified for the round of 16 are highlighted in colour. 

2. Automatic assignment of teams in the knockout phase 

 The teams, including the four best third-placed teams, are assigned automatically. 

3. Multilingualism 

The team names and all other texts can be displayed in English, Spanish, Italian, French, German and 

Dutch. In addition, translations in your native language can be added. 

4. Choice of time zone 

 All time zones of the earth can be selected in a selection field, so that the date and time of the matches 

are displayed in your own local time. 

5. Colored highlighting of the current day's matches 

 The date of the matches that take place today is highlighted in color. 

6. Colored highlighting of missing entries 

 If matches have already taken place in the past few days and no results have been entered, the input fields 

are highlighted in color. 

7. Notice if an invalid tie was entered in the knockout phase 

 The matches in the knock-out phase will be decided by a penalty shoot-out at the latest. If a draw is 

entered for both the match result and the penalty shoot-out, a corresponding note will appear below the 

entry fields. 

8. Display of all cases of direct comparison in the group stage 

 If at the end of the group stage two or three teams are tied in points, a direct comparison takes place in 

which only the matches of these teams against each other are taken into account. 

These evaluations are shown on a separate worksheet. 

9. Easier adaptation to future European Championships 

 The formulas are kept flexible enough to save a lot of work when adapting them for the next European 

Championship. 

 

Hint: 

All formulas refer to the ten Excel files that can be downloaded in addition to this tutorial (ZIP file). 

There may be minor deviations compared to the most recent schedule (e.g. due to later changes in the 

"Language" worksheet). 

1 Structure of the schedule 

In the first step we create the template for the schedule. All fields remain empty at first because they later 

contain a formula that inserts the language-dependent content depending on the language selected. We call 

this first worksheet “EURO”. It looks like this: 

 

The entries that can already be seen here are for illustration purposes only. They will be replaced by formulas 

later. 

The columns from column AA and the rows from row 94 are hidden. 

If you want to skip this step, you will find the finished result in the file named „EURO_Tut_1.xlsx“.  

2 Choice of language 

Before entries can be made anywhere, we need a worksheet with the translations of all team names and all 

labels into the various national languages. We are therefore creating a new worksheet with the name 

"Language". The illustration on the next two pages shows what this worksheet looks like. 

 
Illustration continued on next page 

It contains the 55 nations of UEFA. 

For a particular group (e.g. Group A), we later select four nations by entering four of the consecutive numbers 

(Rank) in the group table, e.g. 14 for Turkey, 2 for Italy, 19 for Wales and 9 for Switzerland (see chapter 4.1). 

This gives us maximum flexibility for future schedules as well. 

In the "my language" column, a user can enter all translations in their own native language and then have a 

match schedule completely in their native language. 

If you want to save yourself the work of creating this worksheet, you will find the finished result in the file 

named „EURO_Tut_2.xlsx“. 

 

The following describes how this Excel sheet is created and how it works. 

First, all texts must be entered in columns B, D, F, G, H, I, J and K. 

Column D will play an important role - see chapter 4. 

A selection field is setup in cell N1:  Menu  „Data  Data Tools  Data Validation“. 

 

We choose the option „List“. We specify the range  $B$3:$B$9  as the source for the list elements. In this 

range we enter the languages: „english, spanish, italian, ...“. 

Now we start with the first formulas. In cell C3 we enter the formula  =IF($N$1=B3,1,0). This means: if the 

language "english" was selected in cell N1, a „1“ appears in C3, otherwise a „0“. We now copy this formula 

down to cell C9 (move the mouse to the lower right corner until the mouse pointer turns into a small black 

cross, then drag it down with the left mouse button pressed). 

A „1“ now appears to the right of the language that was selected in field M1. Unfortunately, however, 

someone can also click cell M1 and then press the "Delete" key. Then no language is selected at all. In this 

case, the language "english" should be selected (default value). 

To make that happen, let's change the formula in cell C3:  

=IF(AND($N$1<>B4,$N$1<>B5,$N$1<>B6,$N$1<>B7,$N$1<>B8,$N$1<>B9),1,0) 

That means:  If none of the other languages is selected, then there should be a „1“ for "english". 

In column E we now enter formulas which copy the texts of the selected language into column E. 

We start with cell E5: 

=IF($C$3,F5,IF($C$4,G5,IF($C$5,H5,IF($C$6,I5,IF($C$7,J5,IF($C$8,K5,IF(L5<>"",L5,F5))))))) 

The formula starts with  =IF($C$3,F5,...   It should actually be there  =IF($C$3=1,F5,... 

If there is a „1“ in C3, the English text from F5 should be copied into column E. Since Excel regards a „1“ as 

TRUE and a „0“ as FALSE, it is sufficient to write IF($C$3, here. 

If there is a „1“ in C3 (C3 is TRUE), the text from F5 is adopted in E5, otherwise there is another IF condition: 

IF($C$4,G5,...   If C4 is 1, the Spanish text is copied from cell G5 to E5, and so on. 

The last part of the formula IF(L5<>"",L5,F5) means: If the ‘my language’ option is selected and there is no text 

in column L, the English text will be used. 

  

We copy the formula in E5 down to cell E160. In the lines where no translations take place, we delete the 

formulas, e.g. in E64, E65, E83, ... The dollar signs in the formula are important: $C$3 should remain 

unchanged in the next line, but F5 should become F6, F7, etc. 

The auxiliary columns B, C and E can be hidden later. 

  

As a first test, we connect the cells E1 to U1 on the worksheet "EURO" and enter the formula   

=Language!$E$84 . The title of our match schedule now appears in the selected language. 

The current status of the schedule can be found in the file named „EURO_Tut_2.xlsx“. 

3 Choice of time zone 

We will create a new worksheet called “TimeZone”. 

 

We enter the formula  =Language!$E$87  in cell A1 and merge cells A1 to F1. 

In column C we enter the currently existing time zones. We start with cell C4. 

In cell H1 we create a selection field that contains exactly this list of time zones (range $C$4:$C$39). The 

additional designations in column D are for convenience only and have no further functionality. 

The numbers in the range $B$4:$B$39 are decisive for the calculations of date and time. These are the shifts in 

hours compared to the coordinated universal time UTC. Central European Time CET, for example, is UTC+1. 

These numbers are not generated by a formula, but entered manually. 

The kick-off times for the individual matches are always given by UEFA in local time. So that this schedule can 

also be easily adapted for future European Championships, we connect cells L5 and L6 and create the same 

selection field here as in H1. The time zone of the host country should be entered here. The columns from 

column K onwards will be hidden later. 

As the EURO 2020/2021 was held in different countries, the kick-off times were given by UEFA in CEST (Central 

European Summer Time). Therefore we select the time zone UTC+2 in this second selection field. 

We will take this time zone into account in the conversion formulas for the various local times. In this way, you 

only need to enter the official kick-off times of the host country for each European Championship and select 

the time zone of the host country in the second selection field. Everything else is then calculated automatically 

(see Section 4.2). 

In column B, the hour shifts are in a form that we can read. Excel saves the hours in fractions of a day in its 

date and time formats. So three hours would be 3/24 = 0.125. 

In Excel, a complete date with time (a timestamp) is nothing more than a decimal number indicating the 

number of days that have passed since 01.01.1900, 00:00 hours. The whole days are in front of the comma, 

the digits after the comma indicate the fraction of a day. So these digits contain the hours, minutes and 

seconds. 

If we add the decimal number 0.125 to a full Excel date, we have added three hours to it. That's exactly what 

we want to achieve when converting to another time zone. 

In cell E4 we enter the formula  =$B4/24  and copy it down to E39. So we have the corresponding decimal 

number for each time zone, which we have to add to a UTC time. 

The cells G4:I6 are merged and we enter the formula  =Language!$E$144. Cell G1 contains the formula 

=Language!$E$140 and the merged cells K5:K6 contain the formula =Language!$E$141. 

We select the cells C4 and D4 and insert a conditional formatting with the formula  =$H$1=$C4. The selected 

time zone should be coloured green. The formatting is transferred to all time zones in columns C and D with 

the help of the "Format painter". 

Columns B and E will be hidden later. 

  

The current status of the schedule can be found in the file named „EURO_Tut_3.xlsx“. 

4 Grouping and entry of the matches 

The simplest way would be to enter the matches in the individual groups directly on the "EURO" worksheet. In 

order to achieve maximum flexibility and make it as convenient as possible to adapt the match schedule for 

future tournaments, we will include two intermediate steps. 

In the first step, we make the grouping on an extra worksheet called "Groups". In the second step, we create 

an overview of the 51 matches of the tournament on a worksheet called "Matches". The transfer to the 

worksheet "EURO" is then done automatically with the help of formulas. 

  

4.1 Grouping 

We create a new worksheet called "Groups". 

The names of the nations are already available in different languages on the "Language" worksheet. The 

grouping should now take place on the worksheet "Groups". 

In column B we enter the abbreviations A1, A2, A3, ... H4. 

In column C we enter the numbers from the list of nations on 

the worksheet "Language", i.e. 14 for Turkey, 2 for Italy, 19 for 

Wales, 9 for Switzerland etc. 

For a new grouping, simply change these numbers. 

In order for the corresponding names to appear automatically 

in column D, we enter the following formula in cell D7: 

=VLOOKUP($C7,Language!$D$5:$E$63,2,0) 

The VLOOKUP function has four parameters. 

1st parameter:   $C7 

The content of C7 is searched for, in our example for 

"14". 

2nd parameter:   Language!$D$5:$E$63 

This is the range where searches are made and found. The 

search column is always the first column on the far left. 

3rd Parameter:   2 

The result is to be found in the second column. 

4th Parameter:   0 

The "0" (or also "FALSE") has the effect that when searching in the search column, only exact matches are 

found. 

So the formula looks in our list of nations to see which name belongs to number 14 and enters this name in 

cell D7. 

We copy the formula down to cell D35. 

4.2 Matches 

We'll create a new worksheet called "Matches". 

 

The idea is this: 

We create an overview of all 51 matches on the "Matches" sheet. We number the matches from 1 to 51. These 

numbers are entered in column B. The order is determined by UEFA's usual scheme for running the 

tournament: first the group matches A1-A2, A3-A4, B1-B2, B3-B4 up to H2-H3, then the knock-out matches. 

On the “EURO” worksheet, our actual schedule, there is a box for each match 

with the venue, date, teams and result. Each of these boxes is given one of 

these numbers at the top left, for example the number 21. The formulas in 

the box then cause the data of match 21 to be displayed in the box. 

When adapting to a new European Championship, we are now free to decide 

where to place match No. 21 on the new schedule. 

In columns C and D of the worksheet "Matches", we first transfer the teams 

according to the usual UEFA scheme.  Match no. 21 would be  E1-E3, i.e. team 1 of group E against team 3 of 

group E. 

In the columns I and J, we do not enter the names directly, but a formula that provides us with the names 

automatically. Cell C4 contains A1. So we have to look up the name of the first team for match no. 1 on the 

sheet "Groups" under A1. We‘ll find this name in cell Groups!D7. 

The VLOOKUP function is available in Excel for such a lookup. Therefore we enter the following formula in cell 

I4 on the "Matches" worksheet:  =VLOOKUP(C4,Groups!$B$7:$D$35,3,0) 

It is important that the first parameter C4 does not contain any dollar signs. So we can copy this formula both 

to the right and down. We go to the lower right corner of cell I4 until the cursor turns into a small black cross 

and drag the formula over the entire range I4: J39. Now the team names appear in this range in the currently 

selected language. 

  

4.3 Kick-off times 

In column E we enter the date and time for each match and format the cells so that the date and time are 

displayed. 

In column F, the times should now be converted into the time zone that was selected in the cell TimeZone!H1 . 

We enter the following formula in F4: 

=E4-VLOOKUP(TimeZone!$L$5,TimeZone!$C$4:$E$39,3,0) 

+VLOOKUP(TimeZone!$H$1,TimeZone!$C$4:$E$39,3,0) 

The first VLOOKUP function looks up the table of all time zones to see which decimal number (positive or 

negative) corresponds to the local time of the host country. 

In the Germany (EURO 2024) example, this is the number 2/24 ≈ 0,0833. 

This number (0.0833) is first subtracted from the time at the venue (E4). Two hours are subtracted. That 

means it is converted to UTC. The expression VLOOKUP(TimeZone!$H$1,TimeZone!$C$4:$E$39,3,0) is now 

used to look up in the table of all time zones which decimal number (positive or negative) still has to be added 

to get the desired local time. 

We copy the formula in F4 down to F58.  

In cells F40, F49, F54 and F57 we delete the formula again. 

4.4 Venues 

It is most convenient if we only have to enter the number of the venue for each match. We look up these 

numbers in the worksheet "Language" in column D and enter them in column G on the worksheet "Matches". 

We now enter the formula  =IFNA(VLOOKUP($G4,Language!$D$97:$E$112,2,0),"")  into cell H4 and copy it 

down to cell H58. In lines 40, 49, 54 and 57 we delete the formula again. The names of the venues now appear 

in column H. 

The IFNA function has two parameters: 

1st parameter: the value that should appear in the cell 

2nd parameter: what is to appear in the cell in the event of the error "#N/A" (no value available). 

In our case, the cell remains empty if the VLOOKUP function finds nothing. 

Columns B, F, H, I and J  should be locked to prevent accidental overwriting. They are no longer changed. 

Columns C, D, E and G, on the other hand, will be readjusted at the next European Championship. 

The current status of the schedule can be found in the file named “EURO_Tut_4.xlsx”. 

5 Matches of the group stage 

We will enter the required formulas for the first group match and then copy these formulas to all other 

matches in the group stage and also in the Round of 16. The formulas for the quarter-finals, semi-finals and 

finals will be discussed and entered later (see chapter 8). 

With the help of the conditional formatting, we will highlight the date fields that contain today's date in color. 

So you can see immediately which matches are taking place today. 

Also with the help of the conditional formatting, we will highlight empty input fields of matches that are in the 

past. So you can see immediately which match results are still to be entered. 

  

5.1 Formulas for the matches 

We go to the "EURO" sheet and fill in the following formulas: 

Cell B9: =Language!$E$93&" A" 

Cell E9: =Language!$E$93&" B" 

etc. until group F. This means that these headings always appear in the selected national language. 

We have numbered all the matches from 1 to 51. We now enter these numbers according to the following 

scheme at the top left of the box with the match data: 

 

Group A Group B Group C Group D Group E Group F 

1 3 5 7 9 11 

2 4 6 8 10 12 

13 15 17 19 21 23 

14 16 18 20 22 24 

25 27 29 31 33 35 

26 28 30 32 34 36 

 

The first match in group A is match no. 1. So we enter a „1“ in cell A11. We enter a „2“ in cell A16. 

Round of 16: 41, 42, 40, 38, 39, 37, 43, 44  (explanation of the sequence in chapter 8) 

Quarter-finals: 45, 46, 47, 48 

Semi-finals: 49, 50 

Finale: 51 

So there is a 41 in cell A52, a 42 in cell D52, a 40 in cell G52 etc. 

Now we are going to fill the boxes for the matches of the group stage and the Round of 16 with content. 

Normally, we would look on the "Matches" sheet to see which data belong to match no. 1 and transfer this 

data to the box. With the help of the VLOOKUP function, we let this transfer of data take place automatically. 

In cell B11 we enter the formula =VLOOKUP(A11,Matches!$B$4:$K$58,7,0). The search criterion in this 

formula is A11. In A11 there is a 1. 

This 1 is searched for on the "Matches" worksheet in the first column of the range B$4:$K$58. After it has 

been found in row 4 of the worksheet, the content in column 7 ("London") is returned by the VLOOKUP 

function. 

“London” now appears as the desired venue in cell B11. 

All other formulas have the same structure: 

Cell B12: =VLOOKUP(A11,Matches!$B$4:$K$58,5,0) 

Cell B13: =VLOOKUP(A11,Matches!$B$4:$K$58,8,0) 

Cell C13: =VLOOKUP(A11,Matches!$B$4:$K$58,9,0) 

A11 is a relative reference (without the dollar sign). That is why we can mark range B11:C13 on the “EURO” 

worksheet and copy it into the boxes for the other group matches. The data of the match with the 

corresponding number appear in each box. We try it out: For example, in cell M16 we change the number 

from 10 to 9. Immediately we see match number 9 in this box. 

This flexibility also makes it easier for us to arrange the matches for the Round of 16 (see Chapter 8). 

  

5.2 Conditional formatting 

We click on the merged cells B12:C12 and select “Start  Conditional Formatting  New Rule” in the menu. We 

select the last option as the rule type and enter the formula =INT(B12)=TODAY(). 

We click on "Format ..." and choose a yellow background. We 

confirm all settings with OK. 

With “Home  Format Painter” we copy this conditional formatting 

to all matches of the group stage and the Round of 16. 

(Double click on "Format Painter" and click on all date fields) 

We click on the connected cells C63:E63 and give these cells the 

same type of conditional formatting, but here we enter the formula 

=INT(C63)=TODAY(). Then we transfer this conditional formatting 

to all date fields of the remaining knockout matches (quarter-finals, 

semi-finals, finals). 

  

The conditional formatting of the input fields for the match results requires four formulas: 

Cell B14: =AND(B12<TODAY(),B14="") 

The conditional formatting is transferred to all input fields on the left for the group stage and the Round of 16, 

i.e. E14, H14, etc. 

Cell C14: =AND(B12<TODAY(),C14="") 

The conditional formatting is transferred to all input fields on the right for the group stage and the Round of 

16, i.e. F14, I14, etc. 

Cell C65: =AND(C63<TODAY(),C65="") 

Transfer to all input fields on the left for the remaining KO matches. 

Cell E65: =AND(C63<TODAY(),E65="") 

Transfer to all input fields on the right for the remaining KO matches. 

  

With the help of the data validation we prevent nonsensical entries for the match results. 

We mark all input fields and select in the menu  Data  Data Tools  Data validation. 

 

We only allow whole numbers between 0 and 99. 

  

The current status of the schedule can be found in the file named “EURO_Tut_5.xlsx”. 

6 Fair play 

In the event that two teams are tied at the end of the group stage in terms of the number of points, the goal 

difference and the number of goals scored and their direct encounter also ended in a draw, the fair play score 

or the European Qualifiers ranking will decide on the better placement. 

In this case, it is sufficient if you enter a bonus point on the beneficiary team, which then brings this team to 

the better position. Alternatively, you can also enter negative values. 

On a new worksheet with the name "FairPlay" we enable this entry of a bonus point. 

 

Cell B2: =Language!$E$86 

Cell B3: =Language!$E$139 

In the cells to the left of the headings - i.e. in A5, A11, A17, D5, D11 and D17 - we enter the letter of the 

relevant group and hide it using the custom formatting ";;;" (three semicolons). 

Cell B5: =Language!$E$93&" "&A5 

Cell C5: =Language!$E$124 

Cell B6: =VLOOKUP(A5&"1",Groups!$B$7:$D$35,3,0) 

Cell B7: =VLOOKUP(A5&"2",Groups!$B$7:$D$35,3,0) 

Cell B8: =VLOOKUP(A5&"3",Groups!$B$7:$D$35,3,0) 

Cell B9: =VLOOKUP(A5&"4",Groups!$B$7:$D$35,3,0) 

The last four formulas read the four team names from the "Groups" worksheet. 

We enter zeros in the cells C6 to C9. 

Now we select the range B5:C9 and copy it five times to the positions B11, B17, E5, E11 and E17. 

The current status of the schedule can be found in the file named “EURO_Tut_6.xlsx”. 



Tutorial: Create EXCEL schedule 19 Last updated: 2023-03-21 

7 Calculation of the group tables 

The aim is to display the current status of the group table in range B3:C6 on our schedule and to automatically 

enter the order of the teams in the final table B46:B49. 

We create a new worksheet and call it "GrpA". It will look like this: 

 

We then create the other worksheets with the names "GrpB", "GrpC", etc. by making seven copies of the 

worksheet "GrpA" and only entering a "B" or a "C" etc. in cell Q1! 

  

7.1 Transfer of the match results 

Consideration: 

We find the values for group A on the schedule in columns B and C, those for group B in columns E and F, and 

so on. That is the only difference. 

Therefore, we first create a new worksheet with the name "References" and create a small table here in the 

range B4:D12 in which we enter the two relevant columns for each group (see illustration). 

On the worksheet "GrpA" we enter the following formulas in cells B1 and D1: 

B1: =VLOOKUP($Q$1,References!$B$5:$D$10,2,0) 

D1: =VLOOKUP($Q$1,References!$B$5:$D$10,3,0) 

A "B" then appears in cell B1 and a "C" in cell D1. The letter in cell Q1 thus determines which two column 

names are in cells B1 and D1. 

The formula in cell B4 could simply be:  =EURO!B13.  But then this formula would only be correct for group A. 

Therefore, we will assemble this formula text with the help of the function INDIRECT. 

If we write  =INDIRECT(A1), in cell A1 the INDIRECT function expects a cell reference such as "Z8", and the 

function would return the cell contents of Z8. 

But if we write  =INDIRECT(„A1“), the INDIRECT function returns the cell contents of A1. 

The formula is identical to  =A1. 

We use this second case for our purposes. In the end, the formula should be identical to   

=INDIRECT(“EURO!B13“). 

We will now assemble the character string within the brackets from individual parts. In Excel formulas, strings 

are concatenated with "&". 

It looks like this: “EURO!“ & $B$1 & “13“  (without the spaces). There is a "B" in  $B$1. 

For the second match in group A, this string would have to be  “EURO!“ & $B$1 & “18“, because the second 

match can be found in line 18. 

To be even more flexible, we create a small table of these line numbers in range  

References!G5:G10 and a small table of line numbers for the match results in 

range  

References!H5:H10  (see figure on the left). 

In our string we replace the "13" at the end with „References!$G5“: 

“‘EURO!“ & $B$1 & References!$G5 

Our formula is now:  =INDIRECT(“EURO!“&$B$1&References!$G5). 

The advantage: We can copy this formula by dragging it down with the mouse, which automatically creates 

$G6, $G7, ... at the end and the line numbers are taken from our small table. 

Small disadvantage: If we change the table name "EURO", the table name is not automatically corrected in 

these formulas because it is within a character string and is just ordinary text for Excel. So that we do not have 

to change all 24 formulas in the “GrpA” worksheet in such a case, we store the table name in a central location 

where we only have to change it once. 

On the worksheet "References", we enter the name "EURO" in cell A1. Our formula must now get this name 

from this cell. The formula then looks like this: 

=INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$G5) .         That is the final formula. 

Additionally, the expression References!$A$1 is placed in inverted commas in this formula. This gives you the 

freedom to choose a name that contains spaces, for example 'EURO 2024' instead of 'EURO'. 

The other three formulas are structured in the same way: 

Cell D4: =INDIRECT("'"&References!$A$1&"'!"&$D$1&References!$G5) 

Cell E4: =INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$H5)          [goals1] 

Cell G4: =INDIRECT("'"&References!$A$1&"'!"&$D$1&References!$H5)          [goals2] 

 So that the number of goals is only taken into account in matches in which both fields for goals are filled in, 

we have to expand the formulas in cells E4 and G4 a little. The final formula says: If the first field is empty or 

the second field is empty, nothing is entered here in cell E4, otherwise the result. 

Cell E4: =IF(OR(goals1=““,goals2=““),““,goals1) 

The terms  INDIRECT(…)  must then be used for "goals1" and "goals2" (see above). 

Final formulas for the goals: 

Cell E4: 

=IF(OR(INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$H5)="",INDIRECT("'"&References!$A$1&"'!

"&$D$1&References!$H5)=""),"",INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$H5)) 

Cell G4: 

=IF(OR(INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$H5)="",INDIRECT("'"&References!$A$1&"'!

"&$D$1&References!$H5)=""),"",INDIRECT("'"&References!$A$1&"'!"&$D$1&References!$H5)) 

Now we are ready: We mark cells B4:G4 and copy the formulas down to row 9. 

To transfer the results of the penalty shootout, we use similar formulas: 

Cell J8: 

=IF(OR(INDIRECT("'"&References!$A$1&"'!"&$B$1&(References!$H9+2))="",INDIRECT("'"&References!$A$1

&"'!"&$D$1&(References!$H9+2))=""),"",INDIRECT("'"&References!$A$1&"'!"&$B$1&(References!$H9+2))) 

The expression  References!$H9+2  is used here for the line number because the penalty shootout results are 

two lines down. 

Cell L8: 

=IF(OR(INDIRECT("'"&References!$A$1&"'!"&$B$1&(References!$H9+2))="",INDIRECT("'"&References!$A$1

&"'!"&$D$1&(References!$H9+2))=""),"",INDIRECT("'"&References!$A$1&"'!"&$D$1&(References!$H9+2))) 

Cell J9: 

=IF(OR(INDIRECT("'"&References!$A$1&"'!"&$B$1&(References!$H10+2))="",INDIRECT("'"&References!$A$

1&"'!"&$D$1&(References!$H10+2))=""),"",INDIRECT("'"&References!$A$1&"'!"&$B$1&(References!$H10+

2))) 

Cell L9: 

=IF(OR(INDIRECT("'"&References!$A$1&"'!"&$B$1&(References!$H10+2))="",INDIRECT("'"&References!$A$

1&"'!"&$D$1&(References!$H10+2))=""),"",INDIRECT("'"&References!$A$1&"'!"&$D$1&(References!$H10+

2))) 

Done! - We have transferred the six matches in Group A and the match results to our “GrpA” worksheet. 

We can now try out the flexibility of these formulas. For example, if we enter the letter "B" in cell Q1, we 

immediately see the matches of group B with their results! 

7.2 Ranking list - part 1 

In cell M4 we enter the formula  =IF(AND($E4<>"",$G4<>""),1,0). Here it is checked whether the match result 

has been entered in full. The formula is copied down to cell M9. In cell M10 we form the sum of the range 

M4:M9. So we have the number of fully registered matches. 

Cell H4: =IF($M4>0,IF($E4>$G4,3,IF($E4=$G4,1,0)),0) 

Cell I4: =IF($M4>0,IF($E4<$G4,3,IF($E4=$G4,1,0)),0) 

The two formulas are copied down. 

In the range H4:H9  3 points appear if team 1 has won the match, 1 point in the case of a tie and 0 if the match 

was lost or no match result was entered. Correspondingly, the points scored by team 2 appear in range I4:I9. 

 

In the range O4:AF7 we are now creating a table with the four teams, in which we calculate a rating number 

for each team that includes all UEFA criteria. This number in column V will then determine the order of the 

four teams. 

In column O is the consecutive number from 1 to 4. In column P are the names of the four teams. 

In cell P4 we enter the formula =VLOOKUP($Q$1&ROW($A1),Groups!$B$7:$D$35,3,0)  and copy it down to 

cell P7. 

The first parameter of the function VLOOKUP is the expression $Q$1 & ROW($A1).  This expression resolves to 

"A1" because there is an "A" in $Q$1 and ROW($A1) returns a "1". The function VLOOKUP looks on the 

worksheet 'Groups' to see which country name can be found under "A1". It then looks in cell P5 for "A2" and 

so on. 

Cell Q4 should contain the number of points that the first team has achieved. To do this, we add all the points 

in column H if the team concerned is in column B:  SUMIF($B$4:$B$9,P4,$H$4:$H$9) . 

In addition, all points in column I are added if the team in question is in column D:   

SUMIF($D$4:$D$9,P4,$I$4:$I$9) 

 The formula in Q4 is therefore: 

=SUMIF($B$4:$B$9,P4,$H$4:$H$9)+SUMIF($D$4:$D$9,P4,$I$4:$I$9) 

The formula can be copied down to Q7. 

In cell R4 we enter the formula for the goal difference:  =S4-T4 . This can also be copied downwards. 

For the goals scored in column S and the goals conceded in column T, formulas with the same structure as for 

the number of points are used. 

Cell S4: =SUMIF($B$4:$B$9,$P4,$E$4:$E$9)+SUMIF($D$4:$D$9,$P4,$G$4:$G$9) 

Cell T4: =SUMIF($B$4:$B$9,$P4,$G$4:$G$9)+SUMIF($D$4:$D$9,$P4,$E$4:$E$9) 

The number of matches won is determined in column U. 

In cell U4 we enter the following formula: 

Cell U4: =COUNTIFS($B$4:$B$9,$P4,$H$4:$H$9,"=3")+ COUNTIFS ($D$4:$D$9,$P4,$I$4:$I$9,"=3") 

The number of times a “3” is entered in columns H and I is counted for each team. 

In the columns W, X, Y, Z and AA the direct comparisons should be taken into account. The formulas for this 

need preparation and will be discussed later (see Chapters 7.5 and 7.6). 

Column AB: 

Column AB should contain the bonus points from the “FairPlay” worksheet. The formula in cell AB4 is: 

=SUMIFS(FairPlayPoints1,FairPlayTeams1,$P4)+SUMIFS(FairPlayPoints2,FairPlayTeams2,$P4) 

We define the four variables FairPlayPoints1, FairPlayTeams1, FairPlayPoints2 and FairPlayTeams2 in the 

Name Manager of Excel (menu Formulas  Name Manager) as follows: 

FairPlayPoints1: FairPlay!$C$6:$C$21 

FairPlayTeams1: FairPlay!$B$6:$B$21 

FairPlayPoints2: FairPlay!$F$6:$F$21 

FairPlayTeams2: FairPlay!$E$6:$E$21 

In columns C and F on the worksheet “FairPlay”, all numbers are added up if the respective team name is to 

the left. 

Column AC: 

In the AC column, it should be displayed for each team whether a match result has already been entered for 

the team in question (TRUE) or not yet (FALSE). If a team has played at least one match, the score and the 

number of goals conceded cannot both be zero. Therefore the formula in AC4 is: 

=(Q4+T4>0) 

This formula is copied down. 

Column AD: 

In the AD column, we calculate a ranking number for each team, taking into account only points, goal 

difference and goals scored. The formula in cell AD4 is: 

=$Q4*FactorPts+(GDzero+$R4)*FactorGD+$S4*FactorFor 

Cells Y11 and Y12: 

In order to take the penalty shootout results into account, we must first check whether the conditions for a 

penalty shootout in either of the last two group matches are even met. Cell Y11 should contain a "1" if these 

conditions for the penultimate group match are met, otherwise a "0". In cell Y12 there should be a "1" if these 

conditions for the last group match are met, otherwise a "0". 

To do this, enter the following formula in cell Y11: 

=IF(AND(VLOOKUP($B$8,$P$4:$AD$7,15,0)=VLOOKUP($D$8,$P$4:$AD$7,15,0),$E$8=$G$8,COUNTIF($Q$4:$

Q$7,VLOOKUP($B$8,$P$4:$Q$7,2,0))=2,$M$10=6),1,0) 

This formula has the following structure: 

=IF(AND(Condition1, Condition 2, Condition 3, Condition 4),1,0) 

The four conditions have the following meaning: 

Condition 1:  VLOOKUP($B$8,$P$4:$AD$7,15,0)=VLOOKUP($D$8,$P$4:$AD$7,15,0) 

The two VLOOKUP functions look in column 15 - which is column AD - to see if the two teams are tied in 

points, goal difference and goals scored. 

Condition 2:  $E$8=$G$8 

The group match in question must end in a draw. 

Condition 3:  COUNTIF($Q$4:$Q$7,VLOOKUP($B$8,$P$4:$Q$7,2,0))=2 

No other team may have the same number of points. In other words, the number of points that the two teams 

have must occur exactly twice. 

The VLOOKUP function determines the number of points of one of the two teams. The COUNTIF function then 

counts how often this number occurs in the range $Q$4:$Q$7. 

Condition 4:  $M$10=6 

Here it is checked whether all match results of the group in question have been entered in full. Because only 

then can the conditions for a penalty shoot-out be checked. 

We enter a corresponding formula in cell Y12: 

=IF(AND(VLOOKUP($B$9,$P$4:$AD$7,15,0)=VLOOKUP($D$9,$P$4:$AD$7,15,0),$E$9=$G$9,COUNTIF($Q$4:$

Q$7,VLOOKUP($B$9,$P$4:$Q$7,2,0))=2,$M$10=6),1,0) 

Using the results in cells Y11 and Y12, we can set up conditional formatting for the “Penalty Shootout” labels 

in cells B35 and B42 on the EURO worksheet. If all the conditions for a penalty shootout are met, they appear 

in red text. 

Column AE: 

We now enter the following formula in cell AE4: 

=IF(OR(AND($Y$11>0,OR(AND($B$8=$P4,$J$8>$L$8),AND($D$8=$P4,$J$8<$L$8))),AND($Y$12>0,OR(AND($B

$9=$P4,$J$9>$L$9),AND($D$9=$P4,$J$9<$L$9)))),1,0) 

Cell AE4 shall have a “1” if the team concerned has won a penalty shootout and all conditions for a penalty 

shootout have been met. Otherwise there should be a "0" in cell AE4. 

Structure of the formula: 

=IF(OR(Penalty shootout won in match 5, Penalty shootout won in match 6)),1,0) 

The part "Penalty shootout won in match 5" consists of: 

AND($Y$11>0,OR(AND($B$8=$P4,$J$8>$L$8),AND($D$8=$P4,$J$8<$L$8))) 

Two conditions must be met here: 

1.  $Y$11>0 

2.  OR(AND($B$8=$P4,$J$8>$L$8),AND($D$8=$P4,$J$8<$L$8)) 

The 1st condition ($Y$11>0) says: 

All requirements for a penalty shootout in the 5th group match are met. 

The 2nd condition says: 

The team in question (e.g. Turkey) is entered in cell B8 ($B$8=$P4) and won the penalty shootout ($J$8>$L$8) 

OR 

the team in question (e.g. Turkey) is entered in cell D8 ($D$8=$P4) and won the penalty shootout ($J$8<$L$8). 

The part "Penalty shootout won in the 6th match" is structured in the same way. 

The formula in cell AE4 is copied down. 

Column AF: 

The ranking in the final ranking list of the European Qualifiers should be transferred to column AF. 

Column D on the ‘Language’ sheet contains the ranking numbers for each team. That these ranks are there in 

ascending order is just cosmetic. They may also be arranged differently, since they are searched for using the 

INDEX and MATCH functions. 

In cell AF4 we enter the following formula: 

=INDEX(Language!$D$5:$D$59,MATCH($P4,Language!$E$5:$E$59,0),1) 

The MATCH($P4,Language!$E$5:$E$59,0) expression returns the team's relative position in the list of all 

teams (column E). The INDEX function then returns the associated rank. 

 

7.3 The Factors sheet 

The sorting of the four teams into the correct ranking is done according to the following principle: In column V 

of the “GrpA” worksheet there is a “ranking number” for each team, which decides on the rank. The team 

with the largest number comes first. 

This number is created in the following way: 

The number of points achieved in column Q has the highest priority. This number of points is therefore 

multiplied by a very large factor. We take the number 10000000. The direct comparisons of three teams in 

column W (see chapter 7.5) has the second highest priority. It is multiplied by the lower factor 1000000 and 

added. The direct comparisons of two teams in column X have the third highest priority. They are multiplied by 

a factor of 100000 and also added. The factors for the remaining criteria (goal difference, goals scored etc.) 

have smaller factors: 1000, 10, etc. 

We store these factors on a separate worksheet called "Factors". There are no formulas here, just texts and 

numbers. It looks like this: 

The individual factors in cells B4 to B15 are given the following variable names in the Name Manager: 

B4: FactorPts (factor points) 

B5: FactorDirC3 (factor direct comparison of 3 teams) 

B6: FactorDirC2 (factor direct comparison of 2 teams) 

B7: FactorGD (factor goal difference) 

B8: FactorFor (factor goals scored) 

B9: FactorPenalty (factor penalty) 

B10: FactorDirC43 (factor direct comparison of 3 teams if 4 teams are tied in points) 

B11: FactorDirC42 (factor direct comparison of 2 teams if 4 teams are tied in points) 

B12: FactorWins (factor wins) 

B13: FactorFairPlay (factor fair play) 

B14: FactorRank (factor ranking) 

B15: FactorRow (factor row) 

Thus, instead of  Factors!$B$4 , we can simply write FactorPts in the formulas. The formulas will then be easier 

to read. 

  

7.4 Ranking list - part 2 

The formulas in W4, X4, Y4, Z4, and AA4 will be discussed later (see Chapters 7.5 and 7.6). 

In cell V4 we enter the following formula: 

=$Q4*FactorPts+(GDzero+$R4)*FactorGD+$S4*FactorFor+$U4*FactorWins+$W4*FactorDirC3+$X4*FactorDi

rC2+$Y4*FactorDirC43+$Z4*FactorDirC42+$AA4*FactorDirC42+$AB4*FactorFairPlay+$AE4*FactorPenalty+(1

00-$AF4)*($M$10>0)*FactorRank 

GDzero is added to the goal difference $R4 so that this value cannot become negative. This formula is still 

incomplete and will be expanded. 

This gives us a complex ranking that takes all of UEFA's criteria into account with the right priority. 

Explanation of the last expression: (100-$AF4)*($M$10>0)*FactorRank 

The number 100-$AF4 increases as the team's $AF4 rank decreases. The team with the rank number 1 has the 

highest possible factor 99. The factor ($M$10>0) becomes 0 (= FALSE) if M10 = 0, ie if no match result has 

been entered yet. In this case, the teams should appear in their original order and not in the order of the 

qualifiers ranking. 

 

Using the calculated ranks in the range V4:V7, we can now sort the group table. The team with the highest 

ranking should come first. In the range O11:V14 we will generate a sorted list of the four teams. 

We enter the following formula in cell V11: 

=LARGE($V$4:$V$7,ROW(A1)) 

and copy it down to V14. 

The formula  =LARGE(Range,1)  returns the largest value of range, 

the formula  =LARGE(Range,2)  the second largest, 

the formula  =LARGE(Range,3)  the third largest, 

the formula  =LARGE(Range,4)  the fourth largest. 

In order to be able to simply copy the first formula in V14 downwards, the numbers 1, 2, 3 and 4 are 

generated automatically. The expression ROW(A1) returns the row number of A1, ie 1. 

When copying down, it becomes ROW(A2), ROW(A3), ROW(A4), so the numbers 2, 3 and 4 are generated. 

In range O11:V14 we now want to create the same group table as in range O4:V7, only this time sorted. To do 

this, the first step is to search for the ranking number of cell V11 in range V4:V7. For example, if it is found in 

cell V5, all values to the left of cell V5 (the values of cells P5:U5) are copied down to cells P11:U11. 

The VLOOKUP function is not suitable here because it can only find values to the right of the lookup column. In 

this case, a combination of the INDEX and MATCH functions helps. 

 First we enter the current number in cells O11 to O14. In cell P11 we enter the following formula: 

=INDEX(P$4:P$7,MATCH($V11,$V$4:$V$7,0)) 

The dollar sign in front of the "P" is missing because the formula is to be copied to the right. Then "P" becomes 

"Q" and so on. 

Explanation of the formula: 

The formula  =INDEX(P$4:P$7,3)  would return the third value from the range P$4:P$7, i.e. the content of cell 

P6. The "3" is now replaced by the expression MATCH($V11,$V$4:$V$7,0). The MATCH function returns the 

correct number because it searches for the content of $V11 in the range $V$4:$V$7 and returns the relative 

position of the value found. The third parameter of the MATCH function is a zero. This means that the function 

is looking for an exact match. 

The formula in P11 is copied to the right and down to U14 and the sorted group table is ready! 

However, the sorted group table still has one error: 

For example, if the second and third team have the same ranking, the MATCH function will only find the 

position of the second team, but no longer the position of the third team. The table then shows the name of 

the second team twice with their values, the third team is missing. So we have to avoid having two or more 

ranks equal. For example, if all four are the same because no match results have yet been entered, the first 

team appears four times in the sorted list. Instead, the four teams should appear in their original order. We 

achieve this by adding the row number multiplied by the factor FactorRow (= 0.00000001) to the ranking 

number. 

The final formula in cell V4 is then: 

=$Q4*FactorPts+(GDzero+$R4)*FactorGD+$S4*FactorFor+$U4*FactorWins+$W4*FactorDirC3+$X4*FactorDi

rC2+$Y4*FactorDirC43+$Z4*FactorDirC42+$AA4*FactorDirC42+$AB4*FactorFairPlay+$AE4*FactorPenalty+(1

00-$AF4)*($M$10>0)*FactorRank+(8-ROW())*FactorRow 

The ROW() function returns the row number of the cell in which the formula is located. The expression 8-

ROW() generates the number 4 for V4, the number 3 for V5, the number 2 for V6 and the number 1 for V7. 

Thus, if all teams are equal, the team in line 4 appears at the top, the team in line 5 appears in the second 

position, and so on. 

7.5 Direct comparison of three teams 

In contrast to the FIFA World Cup, direct comparisons of two or three teams are much more common. This is 

because this is the case at the European Championships if the teams involved have the same number of 

points. At the FIFA World Cup, on the other hand, a direct comparison is only carried out if the teams involved 

have the same number of points, the same goal difference and the same number of goals scored. This is less 

common, but still occasionally occurs. 

One of the reasons why direct comparisons are so important at the UEFA European Championship is that they 

take precedence over goal difference and the number of goals scored. 

In order to carry out a direct comparison of three teams, we create four group tables with three teams each, in 

which only the matches are taken into account that these three teams have played against each other. 

  

For the team names we enter the following formulas: 

Cell P18: =$P$4 

Cell P19: =$P$5 

Cell P20: =$P$6 

Cell P21: =$P$7 

In cell Q18 we enter the following formula: 

=SUMIFS($H$4:$H$9,$B$4:$B$9,$P18,$D$4:$D$9,"<>"&$P$21) 

+SUMIFS($I$4:$I$9,$B$4:$B$9,"<>"&$P$21,$D$4:$D$9,$P18) 

Parameters of the function SUMIFS: 

$H$4:$H$9 Sum range: the range in which the numbers are added under certain conditions 

$B$4:$B$9 Criteria range 1: the range that is checked for criterion 1 

$P18 Criterion 1: the content of $P18 is checked for agreement in criteria range 1 

$D$4:$D$9 Criteria range 2: the range that is checked for criterion 2 

"<>"&$P$21 Criterion 2: in Criteria range 2 we check whether the content of $P18 is not the same as the 

 content of $P$21 

The second function SUMIFS in the formula has the same structure. The points scored for each of the three 

teams are added up if they were not scored against the fourth team in cell P21. 

The formulas in S18 and T18 work on the same principle: 

Cell S18: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$P18,$D$4:$D$9,"<>"&$P$21) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,"<>"&$P$21,$D$4:$D$9,$P18) 

Cell T18: 

=SUMIFS($G$4:$G$9,$B$4:$B$9,$P18,$D$4:$D$9,"<>"&$P$21) 

+SUMIFS($E$4:$E$9,$B$4:$B$9,"<>"&$P$21,$D$4:$D$9,$P18) 

Cell R18: =S18-T18 

Cell V18: =Q18*FactorPts+(GDzero+R18)*FactorGD+S18*FactorFor 

In a direct comparison, the points achieved, the goal difference and the goals scored are compared with each 

other. Cell V18 therefore contains a rating number for direct comparison. 

The formulas from Q18 to V18 are now copied down to line 20. 

In column W we now generate a ranking number for each of the three teams. 

Cell W18: =RANK(V18,V$18:V$20,1) 

The RANK function determines the rank of the number in V18 within the range V$18:V$20. The third 

parameter, the "1", means that the largest number gets the lowest rank, ie rank 3. 

The rank number in cell W18 is now transferred to cell X18 if that team is tied with second and third teams but 

not fourth. Otherwise, a zero is entered in cell X18. This ensures that the rank determined in cell W18 is only 

taken into account if the requirement for a direct comparison is met. 

Cell X18: =IF(AND($Q$4=$Q$5,$Q$4=$Q$6,$Q$4<>$Q$7),W18,0) 

In cell W4, all the numbers in the range X18:X35 are added up if the team name 'Turkey' is in the P column. 

The following formula does this: 

Cell W4: =SUMIFS($X$18:$X$35,$P$18:$P$35,$P4) 

Since each team in the range X18:X35 can only be involved once in a direct comparison of three teams, cell W4 

contains the rank of the direct comparison or a zero. 

The formula in cell W4 is copied down to cell W7. 

We select the range Q18:X18 and copy the formulas down to rows 19 and 20. 

 

The rest of the formulas in range P18:X35 look very similar: 

Group 2 

Cell P23: =$P$4 

Cell P24: =$P$5 

Cell P25: =$P$7 

Cell P26: =$P$6 

Cell Q23: 

=SUMIFS($H$4:$H$9,$B$4:$B$9,$P23,$D$4:$D$9,"<>"&$P$26) 

+SUMIFS($I$4:$I$9,$B$4:$B$9,"<>"&$P$26,$D$4:$D$9,$P23) 

Cell R23: =S23-T23 

Cell S23: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$P23,$D$4:$D$9,"<>"&$P$26) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,"<>"&$P$26,$D$4:$D$9,$P23) 



Tutorial: Create EXCEL schedule 30 Last updated: 2023-03-21 

Cell T23: 

=SUMIFS($G$4:$G$9,$B$4:$B$9,$P23,$D$4:$D$9,"<>"&$P$26) 

+SUMIFS($E$4:$E$9,$B$4:$B$9,"<>"&$P$26,$D$4:$D$9,$P23) 

Cell V23: =Q23*FactorPts+(GDzero+R23)*FactorGD+S23*FactorFor 

Cell W23: =RANK(V23,V$23:V$25,1) 

Cell X23: =IF(AND($Q$4=$Q$5,$Q$4=$Q$7,$Q$4<>$Q$6),W23,0) 

 

Group 3 

Cell P28: =$P$4 

Cell P29: =$P$6 

Cell P30: =$P$7 

Cell P31: =$P$5 

Cell Q28: 

=SUMIFS($H$4:$H$9,$B$4:$B$9,$P28,$D$4:$D$9,"<>"&$P$31) 

+SUMIFS($I$4:$I$9,$B$4:$B$9,"<>"&$P$31,$D$4:$D$9,$P28) 

Cell R28: =S28-T28 

Cell S28: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$P28,$D$4:$D$9,"<>"&$P$31) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,"<>"&$P$31,$D$4:$D$9,$P28) 

Cell T28: 

=SUMIFS($G$4:$G$9,$B$4:$B$9,$P28,$D$4:$D$9,"<>"&$P$31) 

+SUMIFS($E$4:$E$9,$B$4:$B$9,"<>"&$P$31,$D$4:$D$9,$P28) 

Cell V28: =Q28*FactorPts+(GDzero+R28)*FactorGD+S28*FactorFor 

Cell W28: =RANK.GLEICH(V28,V$28:V$30,1) 

Cell X28: =IF(AND($Q$4=$Q$6,$Q$4=$Q$7,$Q$4<>$Q$5),W28,0) 

 

Group 4 

Cell P33: =$P$5 

Cell P34: =$P$6 

Cell P35: =$P$7 

Cell P36: =$P$4 

Cell Q33: 

=SUMIFS($H$4:$H$9,$B$4:$B$9,$P33,$D$4:$D$9,"<>"&$P$36) 

+SUMIFS($I$4:$I$9,$B$4:$B$9,"<>"&$P$36,$D$4:$D$9,$P33) 

Cell R33: =S33-T33 

Cell S33: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$P33,$D$4:$D$9,"<>"&$P$36) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,"<>"&$P$36,$D$4:$D$9,$P33) 

Cell T33: 

=SUMIFS($G$4:$G$9,$B$4:$B$9,$P33,$D$4:$D$9,"<>"&$P$36) 

+SUMIFS($E$4:$E$9,$B$4:$B$9,"<>"&$P$36,$D$4:$D$9,$P33) 

Cell V33: =Q33*FactorPts+(GDzero+R33)*FactorGD+S33*FactorFor 

Cell W33: =RANK(V33,V$33:V$35,1) 

Cell X33: =IF(AND($Q$5=$Q$6,$Q$5=$Q$7,$Q$5<>$Q$4),W33,0) 

 

We enter the following formula in cell W4: 

=SUMIFS($X$18:$X$35,$P$18:$P$35,$P4) 

That means: 

In the range X18:X35  all numbers that belong to the first team ($P4) are added. So if a direct comparison is 

made for this team, a number greater than zero will appear in cell W4. This number is multiplied by the factor 

FactorDirC3 and taken into account in the formula for the rank number (in cell V4) (see Chapter 7.4). 

We copy the formula in W4 down to W7. 

This is not the end of the direct comparison of three teams, because it can happen that two of the three teams 

are still not distinguishable. In this case, another direct comparison of these two teams takes place. 

 

Special case: 

Additional direct comparison of two teams 

In order to simplify the formulas in this case, we copy the match results of direct matches into the range 

AA18:AC35. For example, cells AA18 and AA19 contain the result of the Turkey-Italy match, cells AB18 and 

AB20 contain the result of the Turkey-Wales match, and so on. 

The match results are copied from the range E4:G9. Cell AA18 has the formula =$I$18, cell AA19 has the 

formula =$H$19, and so on. 

In cell Y18 we enter the following formula: 

=IF(AND(V18=V19,V18<>V20),IF(AA18>AA19,3,IF(AA18=AA19,1,0)),IF(AND(V18=V20,V18<>V19),IF(AB18>AB

20,3,IF(AB18=AB20,1,0)),"")) 

In the outer IF function IF(AND(V18=V19,V18<>V20),<Yes>,<No>) it is first checked whether the team in line 

18 (Turkey) is tied on points with the team in line 19 ( Italy) but not tied with team in row 20 (Wales). 

If <Yes>: 

The expression IF(AA18>AA19,3,IF(AA18=AA19,1,0)) means that the score is entered as 3, 1 or 0 depending on 

the match result. 

If <No>: 

The expression IF(AND(V18=V20,V18<>V19),IF(AB18>AB20,3,IF(AB18=AB20,1,0)),"") checks whether the 

team in line 18 (Turkey) is tied with the team in row 20 (Wales) but not tied with the team in row 19 (Italy). If 

so, the score 3, 1, or 0 is entered depending on the match result, if no, the cell is empty. 

Conclusion: 

If there is a head-to-head comparison of the team in row 18 with either of the other two, cell Y18 contains the 

score of the head-to-head (3, 1, or 0). 

The formulas in the remaining cells of column Y read accordingly: 

Cell Y19: 

=IF(AND(V19=V18,V19<>V20),IF(AA19>AA18,3,IF(AA19=AA18,1,0)),IF(AND(V19=V20,V19<>V18),IF(AC19>AC

20,3,IF(AC19=AC20,1,0)),"")) 

Cell Y20: 

=IF(AND(V20=V18,V20<>V19),IF(AB20>AB18,3,IF(AB20=AB18,1,0)),IF(AND(V20=V19,V20<>V18),IF(AC20>AC1

9,3,IF(AC20=AC19,1,0)),"")) 

Cell Y23: 

=IF(AND(V23=V24,V23<>V25),IF(AA23>AA24,3,IF(AA23=AA24,1,0)),IF(AND(V23=V25,V23<>V24),IF(AB23>AB

25,3,IF(AB23=AB25,1,0)),"")) 

Cell Y24: 

=IF(AND(V24=V23,V24<>V25),IF(AA24>AA23,3,IF(AA24=AA23,1,0)),IF(AND(V24=V25,V24<>V23),IF(AC24>AC

25,3,IF(AC24=AC25,1,0)),"")) 

Cell Y25: 

=IF(AND(V25=V23,V25<>V24),IF(AB25>AB23,3,IF(AB25=AB23,1,0)),IF(AND(V25=V24,V25<>V23),IF(AC25>AC2

4,3,IF(AC25=AC24,1,0)),"")) 

Cell Y28: 

=IF(AND(V28=V29,V28<>V30),IF(AA28>AA29,3,IF(AA28=AA29,1,0)),IF(AND(V28=V30,V28<>V29),IF(AB28>AB

30,3,IF(AB28=AB30,1,0)),"")) 

Cell Y29: 

=IF(AND(V29=V28,V29<>V30),IF(AA29>AA28,3,IF(AA29=AA28,1,0)),IF(AND(V29=V30,V29<>V28),IF(AC29>AC

30,3,IF(AC29=AC30,1,0)),"")) 

Cell Y30: 

=IF(AND(V30=V28,V30<>V29),IF(AB30>AB28,3,IF(AB30=AB28,1,0)),IF(AND(V30=V29,V30<>V28),IF(AC30>AC2

9,3,IF(AC30=AC29,1,0)),"")) 

Cell Y33: 

=IF(AND(V33=V34,V33<>V35),IF(AA33>AA34,3,IF(AA33=AA34,1,0)),IF(AND(V33=V35,V33<>V34),IF(AB33>AB

35,3,IF(AB33=AB35,1,0)),"")) 

Cell Y34: 

=IF(AND(V34=V33,V34<>V35),IF(AA34>AA33,3,IF(AA34=AA33,1,0)),IF(AND(V34=V35,V34<>V33),IF(AC34>AC

35,3,IF(AC34=AC35,1,0)),"")) 

Cell Y35: 

=IF(AND(V35=V33,V35<>V34),IF(AB35>AB33,3,IF(AB35=AB33,1,0)),IF(AND(V35=V34,V35<>V33),IF(AC35>AC3

4,3,IF(AC35=AC34,1,0)),"")) 

However, the scores in column Y should only be taken into account if there is already a direct comparison of 

the three teams, i.e. if column X contains a value greater than zero. Therefore we enter the formula 

=IF(X18>0,Y18,"") in cell Z18 and copy it down to cell Z35. 

Column Z therefore contains the scores of the direct comparison of the team in question with another if the 

team in question is involved in a direct comparison of three teams at the same time. 

The results in the range Z18:Z35 are taken into account in the formulas in the range X4:X7 (see Section 7.6). 

The formulas in the X4:X7 range also take into account direct comparisons between two teams in cases where 

there is no direct comparison between three teams. These cases we consider in the next chapter. 

 

7.6 Direct comparison of two teams 

In order to take into account the direct comparison of two teams, we first create a small table with the results 

of the direct encounters in range  E17:K21 . 

 

We transfer the team names to cells E18 to E21 with the formulas =$P$4, =$P$5, =$P$6 and =$P$7. We 

connect cells E18:G18, E19:G19, E20:G20 and E21:G21 respectively. The full names must be here! The first 

three letters of the team names are sufficient for the labeling in cells H17 to K17 above: 

Cell H17: =LEFT($P$4,3) 

Correspondingly in the other cells. 

The number of goals scored by team 1 against team 2 should be in cell I18. Cell H19 then shows the number of 

goals that team 2 scored against team 1. We enter the following formula in I18: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$P$4,$D$4:$D$9,$P$5) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,$P$5,$D$4:$D$9,$P$4) 

In the range $E$4:$E$9, the numbers are added if team 1 is on the left and team 2 is on the right. In the range 

$G$4:$G$9, the numbers are added if team 2 is on the left and team 1 is on the right. This way we get exactly 

the number of goals that team 1 scored against team 2. 

The rest of the formulas work in the same way. They are: 

Cell G18: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$4,$D$4:$D$9,$P$6) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$6,$D$4:$D$9,$P$4) 

Cell H18: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$4,$D$4:$D$9,$P$7) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$7,$D$4:$D$9,$P$4) 

Cell E19: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$5,$D$4:$D$9,$P$4) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$4,$D$4:$D$9,$P$5) 

Cell G19: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$5,$D$4:$D$9,$P$6) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$6,$D$4:$D$9,$P$5) 

Cell H19: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$5,$D$4:$D$9,$P$7) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$7,$D$4:$D$9,$P$5) 

Cell E20: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$6,$D$4:$D$9,$P$4) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$4,$D$4:$D$9,$P$6) 

 Cell F20: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$6,$D$4:$D$9,$P$5) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$5,$D$4:$D$9,$P$6) 

Cell H20: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$6,$D$4:$D$9,$P$7) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$7,$D$4:$D$9,$P$6) 

Cell E21: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$7,$D$4:$D$9,$P$4) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$4,$D$4:$D$9,$P$7) 

Cell F21: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$7,$D$4:$D$9,$P$5) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$5,$D$4:$D$9,$P$7) 

Cell G21: =SUMIFS($E$4:$E$9,$B$4:$B$9,$P$7,$D$4:$D$9,$P$6) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$P$6,$D$4:$D$9,$P$7) 

A direct comparison of two teams takes place when both teams have the same number of points. 

In the range L18:L21, a team is now given a 3 (three points) if it has won a direct comparison. If there is a tie in 

the direct comparison, it receives a 1 (one point), and if it lost in the direct comparison, it receives a 0 (zero 

points). Otherwise these cells remain empty. 

To do this, we enter the following formula in cell L18: 

=IF(AND($Q$4=$Q$5,$Q$4<>$Q$6,$Q$4<>$Q$7),IF(I18>H19,3,IF(I18=H19,1,0)), 

IF(AND($Q$4=$Q$6,$Q$4<>$Q$5,$Q$4<>$Q$7),IF(J18>H20,3,IF(J18=H20,1,0)), 

IF(AND($Q$4=$Q$7,$Q$4<>$Q$5,$Q$4<>$Q$6),IF(K18>H21,3,IF(K18=H21,1,0)),""))) 

The first IF function means: If team 1 is tied with team 2 in points, but otherwise with no other team 

($Q$4=$Q$5,$Q$4<>$Q$6,$Q$4<>$Q$7), then it gets 3 points for a win against team 2 (I18>H19), 1 point for 

a draw (I18=H19) and 0 points for a defeat. 

The second and third IF functions test whether a direct comparison of team 1 with team 3 or team 4 takes 

place. 

The formulas for cells L19, L20 and L21 are accordingly: 

Cell L19: 

=IF(AND($Q$5=$Q$4,$Q$5<>$Q$6,$Q$5<>$Q$7),IF(H19>I18,3,IF(H19=I18,1,0)), 

IF(AND($Q$5=$Q$6,$Q$5<>$Q$4,$Q$5<>$Q$7),IF(J19>I20,3,IF(J19=I20,1,0)), 

IF(AND($Q$5=$Q$7,$Q$5<>$Q$4,$Q$5<>$Q$6),IF(K19>I21,3,IF(K19=I21,1,0)),""))) 

Cell L20: 

=IF(AND($Q$6=$Q$4,$Q$6<>$Q$5,$Q$6<>$Q$7),IF(H20>J18,3,IF(H20=J18,1,0)), 

IF(AND($Q$6=$Q$5,$Q$6<>$Q$4,$Q$6<>$Q$7),IF(I20>J19,3,IF(I20=J19,1,0)), 

IF(AND($Q$6=$Q$7,$Q$6<>$Q$4,$Q$6<>$Q$5),IF(K20>J21,3,IF(K20=J21,1,0)),""))) 

Cell L21: 

=IF(AND($Q$7=$Q$4,$Q$7<>$Q$5,$Q$7<>$Q$6),IF(H21>K18,3,IF(H21=K18,1,0)), 

IF(AND($Q$7=$Q$5,$Q$7<>$Q$4,$Q$7<>$Q$6),IF(I21>K19,3,IF(I21=K19,1,0)), 

IF(AND($Q$7=$Q$6,$Q$7<>$Q$4,$Q$7<>$Q$5),IF(J21>K20,3,IF(J21=K20,1,0)),""))) 

In cell X4 we enter the formula  

=IF($L18="",0,$L18)+SUMIFS($Z$18:$Z$35,$P$18:$P$35,$P4) 

and copy this formula down to X7. 

The first addend  IF($L18="",0,$L18)  contains the number of points from a direct comparison of two teams, 

the second addend  SUMIFS($Z$18:$Z$35,$P$18:$P$35, $P4)  the number of points from the direct 

comparison with another team, if a direct comparison of three teams takes place at the same time. (Only one 

of the two addends can be greater than zero.) 

The number in cell X4 is multiplied by the factor FactorDirC2 and taken into account in the formula for the 

ranking number in cell V4. 

 

7.7 Direct comparison of all four teams 

If all four teams have the same number of points, the four teams are first ranked according to the criteria of a 

direct comparison (points, goal difference and number of goals scored). Various special cases can occur: 

1. One or two further direct comparisons of two teams may be required. 

2. Another direct comparison of three teams may be required. 

3. Another direct comparison of three teams may be required and within this comparison another direct 

comparison of two teams. 

 

We first consider point 1: 

Additional direct comparison of two teams 

In cell AF16 we enter the following formula: 

=AND($Q$4=$Q$5,$Q$4=$Q$6,$Q$4=$Q$7) 

This is a simple test to see if all four teams are tied. As a result, the cell contains the value TRUE or the value 

FALSE. 

In cell M18 we enter the following formula: 

=IF($AF$16,IF(AND($AD$4=$AD$5,$AD$4<>$AD$6,$AD$4<>$AD$7),IF(I18>H19,3,IF(I18=H19,1,0)),IF(AND($A

D$4<>$AD$5,$AD$4=$AD$6,$AD$4<>$AD$7),IF(J18>H20,3,IF(J18=H20,1,0)),IF(AND($AD$4<>$AD$5,$AD$4<

>$AD$6,$AD$4=$AD$7),IF(K18>H21,3,IF(K18=H21,1,0)),""))),"") 

To remember: 

Cells AD4:AD7 contain rating numbers, which only contain the number of points, the goal difference and the 

number of goals scored (see Chapter 7.2). 

The formula in cell M18 starts with IF($AF$16,...). That means: If all four teams are equal on points, then... 

The expression IF(AND($AD$4=$AD$5,$AD$4<>$AD$6,$AD$4<>$AD$7),…) checks if team 1 matches team 2 

in score, goal difference and goals scored agrees, but matches no other team.... 

If so, a direct comparison of Team 1 and Team 2 is due: 

IF(I18>H19,3,IF(I18=H19,1,0)) 

3 points, 1 point or 0 points are entered in cell M18. 

If not, it is checked whether a direct comparison of team 1 with team 3 or team 4 is necessary. If necessary, 

the corresponding number of points from the direct comparison is also entered in cell M18 in these cases. 

The formulas in the cells below read accordingly: 

Cell M19: 

=IF($AF$16,IF(AND($AD$5=$AD$4,$AD$5<>$AD$6,$AD$5<>$AD$7),IF(H19>I18,3,IF(H19=I18,1,0)),IF(AND($A

D$5<>$AD$4,$AD$5=$AD$6,$AD$5<>$AD$7),IF(J19>I20,3,IF(J19=I20,1,0)),IF(AND($AD$5<>$AD$4,$AD$5<>$

AD$6,$AD$5=$AD$7),IF(K19>I21,3,IF(K19=I21,1,0)),""))),"") 

Cell M20: 

=IF($AF$16,IF(AND($AD$6=$AD$4,$AD$6<>$AD$5,$AD$6<>$AD$7),IF(H20>J18,3,IF(H20=J18,1,0)),IF(AND($A

D$6<>$AD$4,$AD$6=$AD$5,$AD$6<>$AD$7),IF(I20>J19,3,IF(I20=J19,1,0)),IF(AND($AD$6<>$AD$4,$AD$6<>$

AD$5,$AD$6=$AD$7),IF(K20>J21,3,IF(K20=J21,1,0)),""))),"") 

Cell M21: 

=IF($AF$16,IF(AND($AD$7=$AD$4,$AD$7<>$AD$5,$AD$7<>$AD$6),IF(H21>K18,3,IF(H21=K18,1,0)),IF(AND($

AD$7<>$AD$4,$AD$7=$AD$5,$AD$7<>$AD$6),IF(I21>K19,3,IF(I21=K19,1,0)),IF(AND($AD$7<>$AD$4,$AD$7<

>$AD$5,$AD$7=$AD$6),IF(J21>K20,3,IF(J21=K20,1,0)),""))),"") 

In cell AA4 we enter the formula =IF($M18="",0,$M18) and copy it down to cell AA7. The scores from the 

range M18:M21 are transferred to cells AA4:AA7 for consideration in the final scores in range V4:V7. 

 

We now consider point 2: 

Additional direct comparison of three teams 

In cell AD18 we enter the following formula: 

=IF(AND($AF$16,$AD$4=$AD$5,$AD$4=$AD$6,$AD$4<>$AD$7),W18,0) 

When all four teams are equal on points ($AF$16) and the top three have the same goal difference and 

number of goals scored ($AD$4=$AD$5,$AD$4=$AD$6,$AD$4<>$AD$7), the rank number from the direct 

comparison of the three teams concerned (see Chapter 7.5) is copied into cell AD18. Otherwise a zero is 

entered. 

The formula is copied down to cell AD20. 

The range AD23:AD25 now contains the formulas in the event that all four teams are tied and teams 1, 2 and 4 

are compared directly. 

In cell AD23 we enter the formula 

=IF(AND($AF$16,$AD$4=$AD$5,$AD$4=$AD$7,$AD$4<>$AD$6),W23,0) 

and copy it down to cell AD25. 

The rest of the formulas look the same: 

In cell AD28 we enter the formula 

=IF(AND($AF$16,$AD$4=$AD$6,$AD$4=$AD$7,$AD$4<>$AD$5),W28,0) 

and copy it down to cell AD30. 

In cell AD33 we enter the formula 

=IF(AND($AF$16,$AD$5=$AD$6,$AD$5=$AD$7,$AD$5<>$AD$4),W33,0) 

and copy it down to cell AD35. 

In order to include these rank numbers (range AD18:AD35) in the final rating numbers (range V4:V7), in cell Y4 

we enter the formula =SUMIFS($AD$18:$AD$35,$P$18:$P$35,$P4) and copy it down to cell Y7. 

The formula sums all numbers in the range AD18:AD35 for the first team (team name in cell P4). A value 

greater than zero can occur at most once for this team. 

 

We now consider point 3: 

Additional direct comparison of three teams plus additional direct comparison of two teams 

Such a case can actually exist - see "Appendix: Special case". However, according to my considerations, this 

additional comparison of two teams should always result in a draw, so that this point 3 does not actually have 

to be dealt with. But you're often surprised by things you didn't think were possible. Since the additional effort 

is very low, we also include the formulas for this case to be on the safe side: 

In cell AE18 we enter the following formula: 

=IF(AD18>0,Y18,"") 

AD18>0 means that the case "four teams with the same number of points and a direct comparison of three 

teams" has occurred. In this case, the already calculated result of the comparison of two teams is copied from 

cell Y18, otherwise the cell remains empty. 

The formula is copied down to cell AD35. 

In cell Z4 we enter the formula =SUMIFS($AE$18:$AE$35,$P$18:$P$35,$P4) and copy it down to cell Z7. 

 

In order to include the results in the three columns of range Y4:AA7 in the final rating numbers (cells V4:V7), 

we must not use the previous weighting factors for direct comparisons of three or two teams (FactorDirC3 or 

FactorDirC2). Because if all four teams are equal on points, they are first sorted by goal difference and number 

of goals scored - that's the first direct comparison, so to speak - and then by the other direct comparisons. 

So we have to create new weighting factors that are smaller than the weighting factors of goal difference and 

number of goals. On the other hand, they must be greater than the fair play weighting factor. 

We choose: 

FactorDirC43 = 0.01 

FactorDirC42 = 0.001 

The penalty shootout weighting factor does not compete with this because there is no penalty shootout if all 

four teams are tied. 

 

7.8 Ranking list - part 3 

We now make five copies of the worksheet GrpA, which we call GrpB, GrpC, ..., GrpF. 

A copy of a worksheet can be created by right-clicking on the name of the worksheet at the bottom, then 

selecting "Move or copy ..." from the context menu. Don't forget to tick the "Create a copy" box! 

  

On each copy, we only need to change one cell: cell Q1. 

 

For group B we enter a "B" here, for group C a "C" and so on. 

Our formulas in cells B1, D1 and in the ranges B4:G9 and P4:P7 have the effect that just by changing cell Q1 

the whole table is filled with the data of the correct group. 

 

7.9 Group table on the schedule 

On the worksheet "EURO" we enter the formula =Language!$E$100 in the cells C2, F2, I2, L2, O2 and R2. The 

spaces in the text  “  Pts.     Goals“  ensure the correct position of the headings "Pts." And "Goals". 

Cell B3: =GrpA!P11 

This formula is copied down to cell B6. 

Cell E3: =GrpB!P11 

This formula is copied down to cell E6. 

We enter the corresponding formulas for the remaining six groups. 

To display the points and the goals, we enter the following formula in cell C3: 

="   "&GrpA!$Q11&"        "&GrpA!$S11&Language!$E$143&GrpA!$T11. 

Here, too, the inserted spaces ensure correct positioning. 

In Language!$E$164 there is a separator for the goal difference (colon or minus) surrounded by a space. 

This formula is also copied down to cell C6. 

In cell F3 we enter the following formula: 

="   "&GrpB!$Q11&"        "&GrpB!$S11&Language!$E$143&GrpB!$T11 

We enter the corresponding formulas for the remaining six groups. 

 

In the Final table in lines 45 to 49, the fields should remain empty if not all matches in the group in question 

have been entered. 

We enter the following formulas in cells B45 and B46: 

Cell B45: =Language!$E$106 

Cell B46: =IF(GrpA!$M$10=6,GrpA!$P11,"") 

The formula in B46 is copied down to cell B49. 

The Final tables of the other groups have corresponding formulas. “GrpA” changes to “GrpB” etc. 

The current status of the schedule can be found in the file named “EURO_Tut_7.xlsx”. 

8 Knockout phase 

8.1 Calculation of the four best third-placed teams 

We create a new worksheet named "ThirdPlaced". It will look like this: 

 

All third-placed teams are listed in the C4:C9 range. For example, cell C4 has the formula =GrpA!P$13. 

In the D4:H9 range, for each team the points achieved, the goal difference, the goals scored, the goals 

conceded and the number of wins are copied from the worksheets of the respective group. For example, the 

formula in cell D4 is =GrpA!Q$13. 

In cell L4 we enter the formula 

=SUMIFS(FairPlayPoints1,FairPlayTeams1,$C4)+SUMIFS(FairPlayPoints2,FairPlayTeams2,$C4) 

and copy it down to cell L9. 

These formulas determine the fair play ratings for each team. 

In cell M4 we enter the formula 

=100-INDEX(Language!$D$5:$D$59,MATCH($C4,Language!$E$5:$E$59,0),1) 

and copy it down to cell M9. 

This should be a number that is higher, the better the rank achieved in the European Qualifiers Ranking. Here 

it is assumed that the rank numbers in column D of the 'Language' worksheet are correct. They don't have to 

be sorted. 

The MATCH($C4,Language!$E$5:$E$59,0) expression returns the position of the team name ($C4) in the 

Language worksheet - for example 19 for Wales. The INDEX function then looks up the rank number to the left 

of the team name. In this case it is 19 again, because the list is sorted here. For a smaller rank number to result 

in a larger rating number, the rank number is subtracted from 100. 

In cell I4 we enter the formula 

=$D4*FactorPts+$E4*FactorGD+$F4*FactorFor+$H4*FactorWins+$L4*FactorFairPlay+$M4*FactorRank 

and copy it down to cell I9. 

Rating numbers are calculated here, taking into account score, goal difference, goals scored, number of wins, 

fair play and European Qualifiers Ranking. Based on these rating numbers, a sorted list of the six third-placed 

teams is now generated in the range B11:J16. The sorting follows the same principle as described in Chap. 7.4. 

The group letters are also sorted in column J, so that it is known from which groups the four best teams come. 

In our case these are the groups F, D, A, C. 

To sort these four alphabetically, in cell L11 we enter the formula  

=SUM(IF($J$11:$J$14<$J11,1))  

and complete the entry with Ctrl-Shift-Enter (Ctrl-Shift-Enter). So that we convert this formula into a so-called 

array formula. In the edit line it appears in curly brackets. We copy the formula down to cell L14. 

For each letter, this formula counts how many smaller ones are in the list. For example, this results in the 

number 3 for the first letter F, because three of the letters are smaller than F. In this way the numbers in the 

range L11:L14 say the order in which the letters are to be arranged. 

In cell M11 we enter the formula 

=INDEX($J$11:$J$14,MATCH(ROW()-11,$L$11:$L$14,0)) 

and copy it down to cell M14. 

This formula sorts the four letters in ascending order. The expression ROW()-11 produces the four numbers 0, 

1, 2 and 3. The expression MATCH(ROW()-11,$L$11:$L$14,0) returns the position of these four numbers in the 

list L11:L14. The INDEX function then looks up the appropriate letter from the list J11:J14. 

The sorted letters are combined into a string in cell J18 using the formula =M11&" "&M12&" "&M13&" 

"&M14. 

The group table in the N11:Q16 range is actually superfluous. It's just a visual convenience. 

The formula in cell J21 is discussed at the end of Chapter 9.1. 

It is:  =IF(AND(TRUNC($I$14,4)=TRUNC($I$15,4),Distinctness!$G$13>0),1,0) 

 

We create a new worksheet named "AssignThird". 

There is no formula on this worksheet. We simply transfer the table for allocation of third-placed teams in the 

round of 16 provided by UEFA. It is important to ensure that there is exactly one space between each letter in 

column C. Because then we can use conditional formatting to color highlight the current group combination. 

The formula for the conditional formatting in line 8 is: 

=$C8=ThirdPlaced!$J$18 

This formatting can be transferred to the other lines with "Format painter". 

The worksheet 'Distinctness' will only be discussed in chapter 9.2. In this chapter, however, we need the 

reference to cell G13. Therefore, we now create a worksheet called 'Distinctness' and enter the following 

formula in cell G13: 

=GrpA!$M$10+GrpB!$M$10+GrpC!$M$10+GrpD!$M$10+GrpE!$M$10+GrpF!$M$10 

The formula returns the total number of all fully entered match results of the group stage. 

The constellation of the third-placed should be displayed in the range T45:Y49 on the 'EURO' worksheet. Cells 

T45 and U45 are merged. Here we enter the formula =Language!$E$126. 

In cell T46 we enter the formula =IF(Distinctness!$G$13>0,ThirdPlaced!$C11,"") and copy it down to cell T49. 

The four qualified third-placed appear here – but only after at least one group match has taken place 

(Distinctness!$G$13>0). 

The two unqualified teams are shown in cells W48 and W49. 

Cell W48: =IF(Distinctness!$G$13>0,ThirdPlaced!$C15,"") 

Cell W49: =IF(Distinctness!$G$13>0,ThirdPlaced!$C16,"") 

To display the points and the goals we enter the formula 

=IF(Distinctness!$G$13>0,"   "&ThirdPlaced!$D11&"        "&ThirdPlaced!$F11&Language!$E$143 

&ThirdPlaced!$G11,"") 

in cell U46 and copy it down to cell U49. 

Cell X48: 

=IF(Distinctness!$G$13>0,"   "&ThirdPlaced!$D15&"        "&ThirdPlaced!$F15&Language!$E$143 

&ThirdPlaced!$G15,"") 

Cell X49: 

=IF(Distinctness!$G$13>0,"   "&ThirdPlaced!$D16&"        "&ThirdPlaced!$F16&Language!$E$143 

&ThirdPlaced!$G16,"") 

Formulas for group membership: 

Cell V46: =IF(Distinctness!$G$13>0,ThirdPlaced!$J11,"") 

The formula is copied down to cell V49. 

Cell Y48: =IF(Distinctness!$G$13>0,ThirdPlaced!$J15,"") 

The formula is copied down to cell Y49. 

Cells W46 and X46 are merged. With the formula =IF(Distinctness!$G$13>0,ThirdPlaced!$J18,"") the current 

group combination is displayed here. 

 

8.2 Round of 16 

We have prepared all match data for the group stage to such an extent that we only need to enter the number 

of the match in the top left next to each match box on the “EURO” worksheet. The match data is then 

automatically taken from the "Matches" sheet. 

We will also apply this principle in the Round of 16. 

There will be an exception for the other matches in the knockout phase (quarter-finals, semi-finals, finals): 

The team names for the individual encounters are not prepared on the "Matches" sheet. 

Actually, it would be best to use this principle for all matches up to the final. But from the quarter-finals 

onwards, the team names depend on entries on the “EURO” worksheet. We would need difficult-to-read 

formulas with unnecessary computational effort to first generate the team names on the "Matches" sheet and 

then transfer them to the "EURO" sheet. 

On the “Matches” worksheet in cells C41 and D41, we see that match number 37 is match 2A against 2B. The 

corresponding team names should now appear in columns I and J - but not statically, but depending on the 

entries in columns C and D - just like in the group matches. 

To do this, we create a small auxiliary table in the range M41:N56. It has the following entries: 

  M N 

41 1A =EURO!$B$46 

42 2A =EURO!$B$47 

43 1B =EURO!$E$46 

44 2B =EURO!$E$47 

45 1C =EURO!$H$46 

46 2C =EURO!$H$47 

47 1D =EURO!$K$46 

48 2D =EURO!$K$47 

49 1E =EURO!$N$46 

50 2E =EURO!$N$47 

51 1F =EURO!$Q$46 

52 2F =EURO!$Q$47 

53 3A/D/E/F *1 

54 3D/E/F *2 

55 3A/B/C/D *3 

56 3A/B/C *4 

 

The team name for 1A is in cell EURO!$B$46, the team name for 2A is in cell EURO!$B$47, and so on. 

The last four formulas are: 

*1: 

=IF(Distinctness!$G$13>0,VLOOKUP(VLOOKUP(ThirdPlaced!$J$18,AssignThird!$C$8:$G$22,2,0), 

CHOOSE({1,2},ThirdPlaced!$J$4:$J$9,ThirdPlaced!$C$4:$C$9),2,0),"") 

*2: 

=IF(Distinctness!$G$13>0,VLOOKUP(VLOOKUP(ThirdPlaced!$J$18,AssignThird!$C$8:$G$22,3,0), 

CHOOSE({1,2},ThirdPlaced!$J$4:$J$9,ThirdPlaced!$C$4:$C$9),2,0),"") 

*3: 

=IF(Distinctness!$G$13>0,VLOOKUP(VLOOKUP(ThirdPlaced!$J$18,AssignThird!$C$8:$G$22,4,0), 

CHOOSE({1,2},ThirdPlaced!$J$4:$J$9,ThirdPlaced!$C$4:$C$9),2,0),"") 

*4: 

=IF(Distinctness!$G$13>0,VLOOKUP(VLOOKUP(ThirdPlaced!$J$18,AssignThird!$C$8:$G$22,5,0), 

CHOOSE({1,2},ThirdPlaced!$J$4:$J$9,ThirdPlaced!$C$4:$C$9),2,0),"") 

These four formulas differ only in a single number (red writing). 

The expression IF(Distinctness!$G$13>0,…) queries whether at least one match result of the group stage has 

been entered. If not, nothing should initially be entered here for the third-placed teams. 

Otherwise, a nested VLOOKUP is performed: 

VLOOKUP(VLOOKUP(ThirdPlaced!$J$18,AssignThird!$C$8:$G$22,2,0), 

CHOOSE({1,2},ThirdPlaced!$J$4:$J$9,ThirdPlaced!$C$4:$C$9),2,0) 

The inner expression VLOOKUP(ThirdPlaced!$J$18,AssignThird!$C$8:$G$22,2,0) reads the current group 

combination (e.g. A C D F) from the cell ThirdPlaced!$J$18 and looks on the worksheet 'AssignThird' to see 

which letter is in column D - here e.g. the letter F. 

For the outer VLOOKUP function, 

the expression CHOOSE({1,2},ThirdPlaced!$J$4:$J$9,ThirdPlaced!$C$4:$C$9) produces a search array 

composed of two columns: ThirdPlaced!$J$4:$J$9 and ThirdPlaced!$C$4:$C$9. In column I the letter F is 

searched and in column C the name 'Portugal' is found. 

If the red 2 in the formula now becomes a 3, the inner VLOOKUP function finds the letter D on sheet 

'AssignThird', and the outer VLOOKUP function finds the name 'Czechia' on sheet 'ThirdPlaced'. 

 

In cell I41 we enter the formula =VLOOKUP(C41,$M$41:$N$56,2,0) and copy it down and to the right over the 

whole range I41:J48. The team names of the Round of 16 matches now appear. 

 The headings "Round of 16 - 1", "Round of 16 - 2", etc. are not officially given. We choose them according to 

the chronological order of the Round of 16 matches. However, the headings can just as well be numbered 

from left to right. 

To make it as easy as possible to adapt the chronological order to future European championship 

tournaments, we enter the numbers from 1 to 8 in the desired order in cells K41 to K48. The rest will be done 

by formulas. 

At the European championship 2020/2021, Match No. 37 (Round of 16 Match 2A-2B) is first, it gets the 

number 1. 

Match No. 38 (Round of 16 Match 1A-2C) is second, it gets the number 2. 

Match No. 39 (Round of 16 Match 1C-3D/E/F) will be the third, it gets the number 3. 

In this way, the chronological order is 1, 2, 3, 4, 5, 6, 7, 8. 

The numbering of the matches (37 - 44) corresponds to the chronological order. This does not always have to 

be the case. 

 

Now we fill the match plan with the dates of the Round of 16. 

On the "EURO" worksheet, we enter the formula   

=VLOOKUP(VLOOKUP(A52,Matches!$B$41:$K$48,10,0),Language!$D$108:$E$115,2,0) in cell B51. 

The inner VLOOKUP function  VLOOKUP(A52,Matches!$B$41:$K$48,10,0)  takes the match number "41" in 

cell A52 and then looks up the number for this match in column K on the worksheet 'Matches'. 

It finds the number "5". 

The outer VLOOKUP function now looks on the worksheet 'Language' in the range $D$108:$E$115 to see 

which heading belongs to the number "5", and finds there the language-dependent heading for the first Round 

of sixteen, in English for example "Round of 16 - 5". The idea of translating only the word "Round of 16" and 

adding a 5 can possibly lead to conflicts with languages where the 5 is not at the end. 

We copy the formulas for range B52:C54 from one of the group matches: for example, mark range B38:C40, 

copy with “Ctrl C”, then click on cell B52 and paste with “Ctrl V”. 

In cell B55 we enter:  =Language!$E$125  ("Penalty shoot-out:"). Cells B56 and C56 are merged. 

They get a conditional formatting (font color red) with the formula =AND(B55=C55,B55<>"",C55<>""). 

In line 58 we enter the following two formulas: 

Cell B58: =VLOOKUP(A52,Matches!$B$4:$K$58,2,0) 

Cell C58: =VLOOKUP(A52,Matches!$B$4:$K$58,3,0) 

Now we mark the range B51:C58 and copy it into the other Round of 16 matches. 

 

Explanation of the order of the Round of 16 matches on the match schedule: 

Final               51 
  

            

1/2-Final       49               50       

1/4-Final   45       46       47       48   

1/8-Final 41   42   40   38   39   37   43   44 

 

If we follow the matches backwards from the final, we can see in the UEFA plan that in the final, the winner of 

match 49 will play against the winner of match 50. In match 49, the winner of match 45 plays against the 

winner of match 46. In match 45, the winner of match 41 plays against the winner of match 42. 

This gives us the following order of the Round of 16 matches: 41, 42, 40, 38, 39, 37, 43, 44. 

 

8.3 Quarter-finals, semi-finals and finals 

For the quarter-finals we enter the numbers 1, 2, 3, 4 in the cells K50:K53 of the worksheet 'Matches'. They 

result from the chronological order of the quarter-finals. Since the order of the semi-finals and the final is 

always the same, we enter the corresponding formulas in this case: 

Cell K55: =Language!$E$120 

Cell K56: =Language!$E$121 

Cell K58: =Language!$E$123 

On the 'EURO' worksheet, we enter the following formulas for the first quarter-final match: 

Cell C61: =VLOOKUP(VLOOKUP(B62,Matches!$B$50:$K$53,10,0),Language!$D$116:$E$119,2,0) 

Cell C62: =VLOOKUP(B62,Matches!$B$4:$K$58,7,0) 

Cell C63: =VLOOKUP(B62,Matches!$B$4:$K$58,5,0) 

Cell C66: =Language!$E$125 

Cell C68: =VLOOKUP(B62,Matches!$B$50:$K$58,2,0) 

Cell E68: =VLOOKUP(B62,Matches!$B$50:$K$58,3,0) 

The cells  C66:E66  are merged and get a conditional formatting for red font colour with the formula 

=AND(C65=E65,C65<>"",E65<>""). 

We mark the range C61:E68 and copy it into the boxes of the remaining quarter-final matches and the first 

semi-final match. The formula for the heading is no longer correct in the semi-final and the error "#N/A" (no 

value available) appears. From the semi-final onwards, we have constant headings with a different formula. In 

cell F71 we therefore enter: 

Cell F71: =VLOOKUP(E72,Matches!$B$50:$K$58,10,0) 

We mark the range F71:H78 and copy it into the boxes of the second semi final and the final. 

 

At this point we have the possibility of further optimization: 

On the 'EURO' spreadsheet, the numbers of the matches in the KO round (round of 16, quarter-finals and 

semi-finals) can be generated automatically using formulas. This has the advantage that once the match plan 

has been filled out on the 'Matches' spreadsheet, all the matches in the KO round are automatically arranged 

correctly. 

On the 'EURO' spreadsheet, cell C68 has the content "W41". This means that the first round of 16 match (far 

left) must be match number 41. We therefore replace the number 41 in cell A52 with the formula 

=RIGHT($C$68,2)*1. 

The last two characters ("41") are truncated from the text "W41". Multiplying by 1 converts the truncated text 

to the number 41. This conversion is necessary because the LOOKUP functions in the range B51:C58 would not 

find the text "41" in the Matches worksheet. They would return the error code #N/A. 

We proceed accordingly with the other match numbers of the KO round. For example, in cell D52 we enter the 

formula =RIGHT($E$68,2)*1. Cell H62 then contains the formula =RIGHT($H$78,2)*1 and cell E72 contains the 

formula =RIGHT($L$88,2)*1. 

Cell K82 contains the number 51, since the final is always match number 51. 

 

Now only the team names are missing. 

In cell C64 we enter the following formula: 

=IF(AND(B55<>"",C55<>""),IF(B55>C55,B54,IF(B55<C55,C54,IF(AND(B57<>"",C57<>""),IF(B57>C57,B54, 

IF(B57<C57,C54,"")),""))),"") 

The winner of match 41 is said to be here. The formula consists of six nested IF functions. The first IF function 

checks whether the match result is entered completely in cells B55 and C55. If not, no team name is entered. 

Otherwise: The next two IF functions check whether one of the two teams has won. In this case, the winner is 

entered. Otherwise (draw): Now it is checked whether the result of the penalty shoot-out is complete and no 

draw has been entered. Accordingly, the winner or nothing is entered for the team name. 

Here are the remaining formulas for the team names: 

Cell E64: 

=IF(AND(E55<>"",F55<>""),IF(E55>F55,E54,IF(E55<F55,F54,IF(AND(E57<>"",F57<>""),IF(E57>F57,E54, 

IF(E57<F57,F54,"")),""))),"") 

Cell I64: =IF(AND(H55<>"",I55<>""),IF(H55>I55,H54,IF(H55<I55,I54,IF(AND(H57<>"",I57<>""),IF(H57>I57,H54, 

IF(H57<I57,I54,"")),""))),"") 

Cell K64: 

=IF(AND(K55<>"",L55<>""),IF(K55>L55,K54,IF(K55<L55,L54,IF(AND(K57<>"",L57<>""),IF(K57>L57,K54, 

IF(K57<L57,L54,"")),""))),"") 

Cell O64: 

=IF(AND(N55<>"",O55<>""),IF(N55>O55,N54,IF(N55<O55,O54,IF(AND(N57<>"",O57<>""),IF(N57>O57,N54, 

IF(N57<O57,O54,"")),""))),"") 

Cell Q64: 

=IF(AND(Q55<>"",R55<>""),IF(Q55>R55,Q54,IF(Q55<R55,R54,IF(AND(Q57<>"",R57<>""),IF(Q57>R57,Q54, 

IF(Q57<R57,R54,"")),""))),"") 

Cell U64: 

=IF(AND(T55<>"",U55<>""),IF(T55>U55,T54,IF(T55<U55,U54,IF(AND(T57<>"",U57<>""),IF(T57>U57,T54, 

IF(T57<U57,U54,"")),""))),"") 

Cell W64: 

=IF(AND(W55<>"",X55<>""),IF(W55>X55,W54,IF(W55<X55,X54,IF(AND(W57<>"",X57<>""),IF(W57>X57, 

W54,IF(W57<X57,X54,"")),""))),"") 

 

Cell F74: 

=IF(AND(C65<>"",E65<>""),IF(C65>E65,C64,IF(C65<E65,E64,IF(AND(C67<>"",E67<>""),IF(C67>E67,C64, 

IF(C67<E67,E64,"")),""))),"") 

Cell H74: 

=IF(AND(I65<>"",K65<>""),IF(I65>K65,I64,IF(I65<K65,K64,IF(AND(I67<>"",K67<>""),IF(I67>K67,I64, 

IF(I67<K67,K64,"")),""))),"") 

Cell R74: 

=IF(AND(O65<>"",Q65<>""),IF(O65>Q65,O64,IF(O65<Q65,Q64,IF(AND(O67<>"",Q67<>""),IF(O67>Q67,O64, 

IF(O67<Q67,Q64,"")),""))),"") 

Cell T74: 

=IF(AND(U65<>"",W65<>""),IF(U65>W65,U64,IF(U65<W65,W64,IF(AND(U67<>"",W67<>""),IF(U67>W67, 

U64,IF(U67<W67,W64,"")),""))),"") 

Cell L84: 

=IF(AND(F75<>"",H75<>""),IF(F75>H75,F74,IF(F75<H75,H74,IF(AND(F77<>"",H77<>""),IF(F77>H77,F74, 

IF(F77<H77,H74,"")),""))),"") 

Cell N84: 

=IF(AND(R75<>"",T75<>""),IF(R75>T75,R74,IF(R75<T75,T74,IF(AND(R77<>"",T77<>""),IF(R77>T77,R74, 

IF(R77<T77,T74,"")),""))),"") 

8.4 Notice when entering an invalid match result 

A draw is not possible in the knockout phase, as the match will definitely be decided by a penalty shoot-out. If 

a tie is entered as a match result, a note should appear under the match result. 

We merge cells B59 and C59 and enter the following formula: 

=IF(AND(B55<>"",C55<>"",B55=C55,B57<>"",C57<>"",B57=C57),Language!$E$138,"") 

We copy this formula into cells E59, H59, K59, N59, Q59, T59 and W59. 

The note appears when the match result and the penalty shoot-out result have been entered in full, but no 

winner emerges. 

We connect cells C69, D69 and E69 and fill in the following formula: 

=IF(AND(C65<>"",E65<>"",C65=E65,C67<>"",E67<>"",C67=E67),Language!$E$138,"") 

We copy this formula into cells I69, O69, U69, F79, R79 and L89. 

 

Finally, the announcement of the new world champion: 

Cell R82: 

=Language!$E$107 

Cell R84: 

=IF(AND(L85<>"",N85<>""),IF(L85>N85,L84,IF(L85<N85,N84,IF(AND(L87<>"",N87<>""),IF(L87>N87,L84, 

IF(L87<N87,N84,"")),""))),"") 

The ranges R82:W83 and R84:W85 are each merged to one cell. 

The match plan is now ready. The rest (chapters 9 and 10) is additional convenience. 

The current status of the schedule can be found in the file named “EURO_Tut_8.xlsx”. 

9 Optional:  The thing with the red dot 

If two or more teams cannot be distinguished at the end of the group stage due to their points, goals and 

direct comparisons, a decision will be made on the basis of the fair play rating or the rank in the European 

Qualifiers rankings. The user of this match plan must then enter additional bonus points on the “FairPlay” 

worksheet. 

So that the occurrence of such a case is clearly recognizable, the affected groups are marked with a red dot. 

The red dot disappears again when the order has been clarified by entering bonus points. 

To make this even more transparent for the user, we also set up an eye-catching red or green message, for 

example 

"Groups with fair play valuation: A, C and F." 

or 

"The placement has been clarified in all groups." 

 

9.1 The red dot 

In cell X11 on the „GrpA“ worksheet we enter the following formula: 

=IF(AND(OR(TRUNC($V$11,4)=TRUNC($V$12,4),TRUNC($V$11,4)=TRUNC($V$13,4),TRUNC($V$11,4)=TRUNC(

$V$14,4),TRUNC($V$12,4)=TRUNC($V$13,4),TRUNC($V$12,4)=TRUNC($V$14,4),TRUNC($V$13,4)=TRUNC($V

$14,4)),$M$10>0),1,0) 

In detail: 

TRUNC($V$11,4) means that the rank number in V11 (e.g. 90055053,0000983) is reduced to four decimal 

places. This cuts off the rank in the European Qualifiers rankings and the line number and leaves the ranking 

number, which contains all UEFA criteria from the number of points to the Fair Play Score. 

If we call these abbreviated ranks R1, R2, R3 and R4, then the formula says: If at least two of these rank 

numbers are the same (R1 = R2 or R1 = R3 or ... R3 = R4) and at least one match result is entered in this group 

($M$10>0), then a 1 appears in cell X11, otherwise a 0. 

If there is a 1 in cell X11, a red dot should be displayed in the relevant group. 

The formula in GrpA!X11 is now copied into the "GrpB", "GrpC" etc. worksheets. 

On the “EURO” worksheet, we merge cells B8 and C8. The red dot should appear here. 

Therefore we enter the formula  =IF(GrpA!$X$11>0,"•","")  here. 

The point can first be created in any cell with “Insert  Symbol” and then inserted into the formula with copy 

and paste. 

On the “EURO” worksheet, the formula in B8 is copied to E8, H8 etc. 

"GrpA" is changed to "GrpB", "GrpC" etc. 

In cell T44 we enter the formula  =IF(ThirdPlaced!$J$21>0,"•",""). 

For third-placed teams, the order of the top four doesn't matter. All that matters is that the right four teams 

are in the top four positions. In order to determine whether the situation is unclear or not, it is therefore 

sufficient to check whether the placements in fourth and fifth place are clearly clarified. 

Therefore, the formula in cell ThirdPlaced!$J$21 is: 

=IF(AND(TRUNC($I$14,4)=TRUNC($I$15,4),Distinctness!$G$13>0),1,0) 

If the rating numbers in $I$14 and $I$15 match in all places up to the fourth decimal place, this means that the 

teams in 4th place and 5th place are equal in points, goal difference, number of goals scored and number of 

wins and no different numbers were entered in the fair play rating either. In this case, the last criterion, the 

position in the European Qualifiers Ranking, would decide and a red dot will be displayed above the list of 

third-placed teams. The additional condition Distinctness!$G$13>0 ensures that the red dot only appears if at 

least one match result has been entered. 

 

9.2 The message 

We now fill out the “Distinctness” worksheet. It should look like this: 

 

In cells B4 to B9 we write the letters from A to F. 

In cells C4 to C9 we enter the formulas  =GrpA!$X$11 ,  =GrpB!$X$11 , … , =GrpF!$X$11. 

This gives us an overview of the groups in which the fair play valuation or the European Qualifiers Rankings will 

decide. These are all groups for which there is a 1 in column C - in the figure above, groups B, C and E. 

In this case, a message should appear on the schedule with the following text: 

Groups with fair play valuation: B, C and E. 

The first part of the message (groups with fair play valuation: ) is a fixed text. The second part (B, C and E.) is 

variable text that looks different depending on the situation. For example, it can also be: B and E. 

To create this second part of the text, we first enter a formula in cell D4 for the number of groups with 

fair play valuation:  =COUNTIF($C$4:$C$9,">0"). 

In column E we enter the letter of the group if the group is not clear. 

Cell E4: =IF($C4>0,$B4,"") 

The formula is copied down to E9. 

We enter the following formulas in cells F4 and G4: 

Cell F4: =LEN($E4) 

Cell G4: =SUM($F$4:$F4) 

and copy both formulas down to line 9. It is important that there is no dollar sign in front of the last 4 in the 

expression $F$4:$F4. When copying, $F4 becomes $F5, etc. 

Side note: Column F is only an auxiliary column to avoid the use of the SUMPRODUCT function, which behaves like an array formula and is therefore 

more difficult to understand. You could also omit column F and enter the formula  = SUMPRODUCT(LEN($E$4:$E4))  in column G. 

The numbers in column G now tell us the position of the three ambiguous groups in the list of these groups. 

The B in line 5 is in the first position. At the 3 in cell G9 we can see that there are two more ambiguous groups. 

Therefore, a comma must follow after the B. The D in line 7 is in the second position. At the 3 in cell G9 we see 

that this is the penultimate position and must be followed by an “and” after the D. The E in line 8 is in the third 

position. At the 3 in cell G9 we can see that this is the last position. Nothing more has to follow the E. The final 

point is placed by the formula in cell J4. 

This logic is summarized in the formulas in column H. 

Cell H4: =IF($E4<>"",$E4&IF($G4=$G$9,"",IF($G4<$G$9-1,", ",Language!$E$137)),"") 

The formula is copied down. The expression Language!$E$137 is the word “and” in the selected language. 

 In cell H11 we put all parts together into one text: 

Cell H11: =$H$4&$H$5&$H$6&$H$7&$H$8&$H$9 

The entire message is then in cell J4:   =Language!$E$135&"  "&$H$11&"." 

Formula in cell J5:   =Language!$E$136 

The message in J5 appears when there is a zero in D4. 

  

Line 7 of the “EURO” worksheet is intended for the message. We merge cells B7:R7 into one cell and center its 

contents. 

In cell B7 we enter the following formula: 

=IF(Distinctness!$G$13=0,"",IF(Distinctness!$D$4>0,Distinctness!$J$4,Distinctness!$J$5)) 

The first condition Distinctness!$G$13=0 means that no message appears if no results have yet been entered 

in the schedule. 

Cell B7 is given a green font. With the help of the conditional formatting, the font color changes to red if there 

is an unclear order of precedence in one of the groups. The formula for the condition is: 

=Distinctness!$D$4>0 . 

The current status of the schedule can be found in the file named “EURO_Tut_9.xlsx”. 

10 Optional:  Additional overview of the direct comparisons 

This overview is intended to show the comparison table of the three teams concerned, when comparing three 

teams directly, in order to make it clear how the ranking is established in this case. The direct comparison of 

two teams is also shown in this overview. 

We create a new worksheet named "Direct comparisons". It should look like this: 

 

Columns A to X are the area visible to the user. Columns Y to AI are hidden. 

  

We will develop the formulas for group A and then copy the entire range B3:AI9 seven times. In each of these 

copies we then only need to adapt the formula for the heading (cells C11, C19 etc.) and to replace the 

character string “GrpA” with “GrpB” or “GrpC” etc. in the formulas by Find/Replace. Therefore, the setting or 

omission of the dollar signs must be carefully observed here. 

10.1 Visible area - part 1 

 

We start with the heading: 

Cell B1: =Language!$E$85 

The cells B1 to W1 are merged. 

For the headings in line 3 we enter the following formulas: 

Cell C3: =Language!$E$93&" A" 

Cell D3: =Language!$E$101 

Cell J3: =Language!$E$104 

Cell P3: =Language!$E$105 

For the headings in line 4 we enter the following formulas: 

Cell D4: =Language!$E$95 

Cell E4: =Language!$E$96 

Cell F4: =Language!$E$97 

Cell I4: =Language!$E$98 

Cell L4: =Language!$E$95 

Cell M4: =Language!$E$99 

Cell P4: =Language!$E$95 

Cell Q4: =Language!$E$96 

Cell R4: =Language!$E$97 

Cell V4: =Language!$E$102 

Cell W4: =Language!$E$103 

We enter the current number in cells B5 to B8. We copy the team names in sorted order from the worksheet 

"GrpA": 

Cell C5: =GrpA!$P$11 

Cell C6: =GrpA!$P$12 

Cell C7: =GrpA!$P$13 

Cell C8: =GrpA!$P$14 

For example, for later copying of the complete group, it is important that there is a dollar sign in front of the 

numbers 11, 12, 13 and 14. 

 In range D5:I8 you should first see a copy of the group table: 

Cell D5: =GrpA!$Q$11 

Cell D6: =GrpA!$Q$12 

Cell D7: =GrpA!$Q$13 

Cell D8: =GrpA!$Q$14 

Cell E5: =GrpA!$R$11 

Cell E6: =GrpA!$R$12 

Cell E7: =GrpA!$R$13 

Cell E8: =GrpA!$R$14 

Cell F5: =GrpA!$S$11 

Cell F6: =GrpA!$S$12 

Cell F7: =GrpA!$S$13 

Cell F8: =GrpA!$S$14 

Cell H5: =GrpA!$T$11 

Cell H6: =GrpA!$T$12 

Cell H7: =GrpA!$T$13 

Cell H8: =GrpA!$T$14 

 

In the range L5:L8 the points of a direct comparison of two teams should be displayed. If there is no direct 

comparison between two teams, the fields remain empty. 

In order to determine whether there is a direct comparison for the team in question with another team, we 

first add further formulas to the group tables. 

On the "GrpA" worksheet, we enter the formula =VLOOKUP($P11,$E$18:$M$21,8,0)<>"" in cell Z11 and copy 

it down to cell Z14. This formula looks in cells L18:L21 to see if there is a direct comparison between two 

teams for the team in question. 'TRUE' or 'FALSE' is entered accordingly. 

In cell AA11 we enter the formula =VLOOKUP($P11,$E$18:$M$21,9,0)<>"" and copy it down to cell AA14. The 

formula looks in cells M18:M21 to see if there is a direct comparison between two teams for the team in 

question in the case of four teams tied on points. 

In cell AB11 we enter the formula =SUMPRODUCT(1*(LEN($Z$18:$Z$35)>0)*($P$18:$P$35=$P11))>0 and 

copy it down to cell AB14. The formula checks whether there is a direct comparison of two teams for the team 

in question in the case of three teams with the same number of points. 

The SUMPRODUCT function runs through the specified ranges $Z$18:$Z$35 and $P$18:$P$35 step by step. 

The expressions 

1*(LEN($Z$18)>0)*($P$18=$P11) 

1*( LEN($Z$19)>0)*($P$19=$P11) 

. . . 

1*( LEN($Z$35)>0)*($P$35=$P11) 

are therefore calculated and the sum of the individual results is formed. 

For example, if cell Z19 contains a number (3, 1, or 0), the expression LEN($Z$19)>0 is TRUE (or 1). Since the 

expression $P$19=$P11 is also TRUE (both cells contain 'Italy'), the product is 1*TRUE*TRUE = 1. The end 

result SUMPRODUCT(. . .)>0 is then also TRUE. 

The next formula in cell AC11 works on the same principle: 

=SUMPRODUCT(1*(LEN($AE$18:$AE$35)>0)*($P$18:$P$35=$P11))>0 

It checks whether there is a direct comparison of two teams for the team in question in the rare case that all 

four teams are equal in points and another direct comparison of two teams is carried out within the direct 

comparison of three teams. 

 

In cell AD11 we enter the formula 

=IF($Z11,VLOOKUP($P11,$E$18:$M$21,8,0),IF($AA11,VLOOKUP($P11,$E$18:$M$21,9,0),IF($AB11,SUMIFS($

Z$18:$Z$35,$P$18:$P$35,$P11),IF($AC11, SUMIFS($AE$18:$AE$35,$P$18:$P$35,$P11),"")))) 

and copy it down to cell AD14. 

There are four possible cases for a direct comparison of two teams. The formulas in cells Z11, AA11, AB11 and 

AC11 test for any of these cases. The formula in cell AD11 now checks whether one of these four cases has 

occurred. If so, it enters the result of the direct comparison in cell AD11. 

It starts with IF($Z11,VLOOKUP($P11,$E$18:$M$21,8,0),… 

So if a direct comparison of two teams is entered in the range L18:L21, it copies this value into cell AD11. 

Otherwise, the expression IF($AA11,VLOOKUP($P11,$E$18:$M$21,9,0),... will check the next case, and so on. 

The results in the range AD11:AD14 are transferred to the 'Direct comparisons' worksheet. On the sheet 

'Direct comparisons' we enter the formula =GrpA!$AD$11 in cell L5 and copy it down to cell L8. 

 

Cells M5:M8 of the 'Direct comparisons' worksheet should show whether a penalty shoot-out in the last group 

match decided the order of the teams. Here, too, the results are first calculated on the group worksheet and 

then transferred to cells M5:M8. 

On the 'GrpA' worksheet in cell AE11 we enter the formula  

=IF(OR(AND($Y$11>0,OR($B$8=$P11,$D$8=$P11)),AND($Y$12>0,OR($B$9=$P11,$D$9=$P11))),SUMIF($B$8:

$B$9,$P11,$J$8:$J$9)+SUMIF($D$8:$D$9,$P11,$L$8:$L$9),"") 

and copy it down to cell AE14. 

The result of the penalty shoot-out is shown here if the team in question is involved. 

The formula has the following structure: 

=IF(OR(Condition1, Condition2),Result,"") 

Condition1: AND($Y$11>0,OR($B$8=$P11,$D$8=$P11)) 

Condition2: AND($Y$12>0,OR($B$9=$P11,$D$9=$P11)) 

Result: SUMIF($B$8:$B$9,$P11,$J$8:$J$9)+SUMIF($D$8:$D$9,$P11,$L$8:$L$9) 

$Y$11>0  means: There will be a penalty shoot-out in the fifth group match (explanation in Chapter 7.2). 

OR($B$8=$P11,$D$8=$P11) means: The team in question ($P11) is involved. 

The SUMIF function adds the numbers in the ranges J8:J9 and L8:L9 if the team in question ($P11) occurs in 

range B8:B9 or D8:D9. 

On the sheet 'Direct comparisons' we enter the formula =GrpA!$AE$11 in cell M5 and copy it down to cell M8. 

This will show the results of a penalty shootout in the last group match in the M5:M8 range. 

 

In range P5:T8, the group table should be displayed for a direct comparison of three teams. To do this, 

additional formulas must first be prepared in the hidden area (see Chapter 10.2). We will enter and discuss 

these formulas later (see Chapter 10.3). 

In range V5:V8 it should be shown whether a bonus for fair play has already been entered: 

Cell V5: 

=SUMIFS(FairPlayPoints1,FairPlayTeams1,$C5)+SUMIFS(FairPlayPoints2,FairPlayTeams2,$C5) 

The formula is copied down to cell V8. 

A red dot should appear in the range W5:W8 if the positioning for the team in question is not yet clear. We 

will enter and discuss the formulas for this later (see Chapter 10.3). 

10.2 Hidden area 

 

In order to show the direct comparisons of three teams, we need a preparation in the hidden area 

(columns Y to AI). 

Four different groups of three are possible in each group. The tables for these four groups of three can be 

found on the “GrpA” worksheet in range O17:AD36. If there is no direct comparison between three teams, the 

X column and the AD column in these four tables are all zeros. Only in the case of a direct comparison of three 

teams do numbers not equal to zero appear in the X column or in the AD column in one of the four tables. We 

want to copy exactly this table into the range AF5:AI7 of our worksheet "Direct comparisons". 

To do this, we need to know which are the correct line numbers. 

We enter the following formula in cell AA5: 

=IF(VLOOKUP($C5,GrpA!$P$18:$X$21,9,0)+VLOOKUP($C5,GrpA!$P$18:$AD$21,15,0)>0, 

MATCH($C5,GrpA!$P$18:$P$21,0),0) 

The expression VLOOKUP($C5,GrpA!$P$18:$X$21,9,0)  looks for the relevant team ($C5) on the worksheet 

"GrpA" in column X to see whether there is a zero (not a direct Comparison) or a number greater than zero 

(direct comparison). 

The expression VLOOKUP($C5,GrpA!$P$18:$AD$21,15,0) looks up the AD column accordingly. 

The expression MATCH($C5,GrpA!$P$18:$P$21,0),0)  tells us at which position the respective team ($C5) is in 

the range GrpA!$P$18:$P$21. 

The formula in AA5 therefore returns the position of the team in question within the first of the four possible 

groups of three if a direct comparison is available, otherwise a zero. 

In the same way, the formulas in cells AB5, AC5 and AD5 return the team's position in the other three possible 

groups of three. These formulas are: 

Cell AB5: 

=I(VLOOKUP($C5,GrpA!$P$23:$X$26,9,0)+VLOOKUP($C5,GrpA!$P$23:$AD$26,15,0)>0, 

MATCH($C5,GrpA!$P$23:$P$26,0),0) 

Cell AC5: 

=I(VLOOKUP($C5,GrpA!$P$28:$X$31,9,0)+VLOOKUP($C5,GrpA!$P$28:$AD$31,15,0)>0, 

MATCH($C5,GrpA!$P$28:$P$31,0),0) 

Cell AD5: 

=I(VLOOKUP($C5,GrpA!$P$33:$X$36,9,0)+VLOOKUP($C5,GrpA!$P$33:$AD$36,15,0)>0, 

MATCH($C5,GrpA!$P$33:$P$36,0),0) 

We copy the formulas in cells AA5, AB5, AC5 and AD5 down to row 8. 

  

In line 9 we form the sum of the respective column: 

Cell AA9: =SUM(AA5:AA8) 

Cell AB9: =SUM(AB5:AB8) 

Cell AC9: =SUM(AC5:AC8) 

Cell AD9: =SUM(AD5:AD8) 

Those four numbers are then either all zero or one of them is a six. 

  

We enter the following formula in cell AE9: 

{=MAX(IF($AA9:$AD9>0,COLUMN($AA9:$AD9)-COLUMN(Z$1),0))} 

This is a so-called array formula. The curly brackets are not allowed to be entered. But you have to finish 

entering the formula with Ctrl-Shift-Enter. Then the curly brackets are set automatically. 

This array formula calculates the four expressions 

IF($AA9>0,COLUMN($AA9)-COLUMN(Z$1),0) 

IF($AB9>0,COLUMN($AB9)-COLUMN(Z$1),0) 

IF($AC9>0,COLUMN($AC9)-COLUMN(Z$1),0) 

IF($AD9>0,COLUMN($AD9)-COLUMN(Z$1),0) 

and returns the maximum of it. 

For example, in the following figure we see a six in cell AB9. 

 

The first, third and fourth expressions result in 0. 

Only for the second expression, the condition  $AB9>0  is true, so that the expression  COLUMN($AB9)-

COLUMN(Z$1)  is calculated here. This expression gives 28 - 26 = 2. The maximum of the four numbers 0, 0, 2, 

0 is then 2. 

The number in cell AE9 tells us that in this example, column 2 of the small table contains line numbers 

greater than 0. 

So we know that we can find the positions or the relative row numbers of the teams in column 2 of the small 

table AA5:AD8. We will use this in the formulas in the P5:T8 range. 

We also know that we have to copy the second table from the “GrpA” worksheet from the four tables in range 

Q18:T36 to the “Direct comparisons” worksheet. And that's what we're going to do next. We use an array 

formula here again, this time not in a single cell, but in a range. 

To do this, we mark the range AF5:AI7 on the "Direct comparisons" worksheet and enter the following formula 

in the editing line above: 

=IF($AA9>0,GrpA!$Q$18:$T$20,IF($AB9>0,GrpA!$Q$23:$T$25,IF($AC9>0,GrpA!$Q$28:$T$30, 

IF($AD9>0,GrpA!$Q$33:$T$35,"0")))) 

The entry must be completed with Ctrl-Shift-Enter. The formula is then automatically enclosed in curly 

brackets.  The formula causes the second table in our example to be copied to the "Direct comparisons" 

worksheet. 

The formula consists of nested IF functions. If $AA9>0, the range  GrpA!$Q$18:$T$20  should be copied. If 

$AB9>0, the range  GrpA!$Q$23:$T$25  should be copied etc.  Otherwise "0" is entered everywhere. 

With the help of the array formula, we only need to enter a single formula for the entire range 

(the so-called array). 

The alternative would be to enter a formula of the type  

=IF($AA9>0,GrpA!$Q$18,IF($AB9>0,GrpA!$Q$23,IF($AC9>0,GrpA!$Q$28,IF($AD9>0,GrpA!$Q$33,"0")))) 

in everyone of the 12 cells. 

 

On the ‘Direct comparisons’ worksheet, we want to put a red dot in cells W5 to W8 for each team with an 

unclear rank. In preparation for this, we first create zeros or ones in cells Y5 to Y8. We enter the following 

formulas in cells Y5 to Y8: 

Cell Y5: 

=IF(AND(OR(TRUNC(GrpA!$V$11,4)=TRUNC(GrpA!$V$12,4),TRUNC(GrpA!$V$11,4)=TRUNC(GrpA!$V$13,4),T

RUNC(GrpA!$V$11,4)=TRUNC(GrpA!$V$14,4)),VLOOKUP(GrpA!$P$11,GrpA!$P$4:$AC$7,14,0)),1,0) 

Cell Y6: 

=IF(AND(OR(TRUNC(GrpA!$V$12,4)=TRUNC(GrpA!$V$11,4),TRUNC(GrpA!$V$12,4)=TRUNC(GrpA!$V$13,4),T

RUNC(GrpA!$V$12,4)=TRUNC(GrpA!$V$14,4)),VLOOKUP(GrpA!$P$12,GrpA!$P$4:$AC$7,14,0)),1,0) 

Cell Y7: 

=IF(AND(OR(TRUNC(GrpA!$V$13,4)=TRUNC(GrpA!$V$11,4),TRUNC(GrpA!$V$13,4)=TRUNC(GrpA!$V$12,4),T

RUNC(GrpA!$V$13,4)=TRUNC(GrpA!$V$14,4)),VLOOKUP(GrpA!$P$13,GrpA!$P$4:$AC$7,14,0)),1,0) 

Cell Y8: 

=IF(AND(OR(TRUNC(GrpA!$V$14,4)=TRUNC(GrpA!$V$11,4),TRUNC(GrpA!$V$14,4)=TRUNC(GrpA!$V$12,4),T

RUNC(GrpA!$V$14,4)=TRUNC(GrpA!$V$13,4)),VLOOKUP(GrpA!$P$14,GrpA!$P$4:$AC$7,14,0)),1,0) 

If the ranking of the team in question, reduced to four decimal places, matches the reduced ranking of another 

team, a 1 is entered, otherwise a 0. 

The AND function AND(OR(…),VLOOKUP(GrpA!$P$11,GrpA!$P$4:$AC$7,14,0)) has the effect that a „1“ is only 

entered if at least one match has been entered for the team in question. 

The expression VLOOKUP(GrpA!$P$11,GrpA!$P$4:$AC$7,14,0) ) reads a Boolean value from column AC of the 

worksheet "GrpA" that indicates whether the number of matches entered for the relevant Crew is greater than 

zero. 

With these formulas, too, it is important that the numbers 11, 12, 13 and 14 are preceded by the dollar sign! 

  

10.3 Visible area - part 2 

In cell W5 we enter the formula  =IF($Y5>0,"•","")  and copy it down to cell W8. 

The special character can be entered in any cell (Insert  Symbol) and then inserted into the editing line with 

Copy + Paste. 

A red dot appears for the teams with an unclear ranking. 

  

In the range P5:T8 the table for the direct comparison of three teams should be shown. 

In cell P5 we enter the following formula: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,1,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,1,$AE9),1))) 

Explanation of the formula: 

IF($AE9=0,"" means: if there is no direct comparison of three teams, nothing is entered. 

The INDEX function in the form INDEX (range, row, column) returns the content of a cell from the specified 

range. The relative positions of the cell in this range are entered in the parameters row and column. For 

example, the expression INDEX(C4:H9,2,3) would return the contents of cell E5. 

If there is a direct comparison, there is a number greater than zero in cell AE9 and in the small table AA5:AD8 

one of the four columns contains the row numbers of the three teams involved in the direct comparison. In 

cell AE9 is the number of the column in which we can find these line numbers. 

The expression IF(INDEX($AA5:$AD8,1,$AE9)=0  therefore checks whether the line number in the first line is 

equal to 0. If this is the case, then it is the team that does not take part in the direct comparison, and nothing 

is entered. Otherwise, the expression INDEX($AF5:$AI7,INDEX($AA5:$AD8,1,$AE9),1) is evaluated and the 

result is returned. 

In this expression we have two nested INDEX functions. 

The inner INDEX function  INDEX($AA5:$AD8,1,$AE9)  gives us the relative row number of the team in 

question, the outer INDEX function gives us the score of the team in question from column 1 of table AF5:AI7. 

If we insert a “2” instead of a “1” for the column number, we get the goal difference and with a “3” or a “4” 

the goals. We use this for the formulas in cells Q5, R5 and T5. They are: 

 

Cell Q5: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,1,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,1,$AE9),2))) 

Cell R5: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,1,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,1,$AE9),3))) 

Cell T5: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,1,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,1,$AE9),4))) 

The remaining formulas in table M5:Q8 all look very similar: 

Cell P6: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,2,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,2,$AE9),1))) 

Cell Q6: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,2,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,2,$AE9),2))) 

Cell R6: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,2,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,2,$AE9),3))) 

Cell T6: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,2,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,2,$AE9),4))) 

Cell P7: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,3,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,3,$AE9),1))) 

Cell Q7: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,3,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,3,$AE9),2))) 

Cell R7: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,3,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,3,$AE9),3))) 

Cell T7: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,3,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,3,$AE9),4))) 

Cell P8: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,4,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,4,$AE9),1))) 

Cell Q8: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,4,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,4,$AE9),2))) 

Cell R8: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,4,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,4,$AE9),3))) 

Cell T8: 

=IF($AE9=0,"",IF(INDEX($AA5:$AD8,4,$AE9)=0,"",INDEX($AF5:$AI7,INDEX($AA5:$AD8,4,$AE9),4))) 

 

A large blue dot should appear to the left of the group name so that the groups in which a direct comparison is 

carried out are even more visible. In cell B3 we enter the following formula: 

=IF(OR($AE9>0,$L5<>"",$L6<>"",$L7<>""),"","") 

The special character can be entered again in any cell with Insert  Symbol and then inserted into the editing 

line with Copy + Paste. It may appear as a space in the editing line, but it still works. 

We now mark the range B3:AI9 and copy it with Ctrl C, for example. 

Then we click on cell B11 and paste the copy, for example with Ctrl V. 

Important: The new range is still marked! 

With Find and Replace we change the text „GrpA" to „GrpB" in all formulas of the marked range. 

In cell C11 we change the formula  =Language!$E$93&" B". 

We create the ranges for the remaining groups in the same way. - All done! 

The final schedule can be found in the file "EURO_Tut_final.xlsx". 

11 Adaptation of the schedule for a new European Championship 

The schedule is designed in such a way that it can be adapted to the next tournament as easily and user-

friendly as possible. 

If you want to be precise, update the rank numbers on the 'Language' worksheet in column D according to the 

new rank order at the end of the qualification phase. No lines need to be swapped! The rank numbers can also 

be unordered - the main thing is that they are unique. However, you can save yourself this effort, because you 

can always clarify an unclear order in the group phase by entering a fair play bonus. 

The following steps are to be carried out in detail: 

 

11.1 Worksheet 'Language' 

Only the venues need to be updated here. 

 

11.2 Local time of the host country 

On the 'TimeZone' worksheet in cell L5, the host country time zone is updated. To do this, the sheet protection 

must be removed and the hidden columns shown. 

 

11.3 Allocation of the teams to the groups 

The 'Groups' worksheet is shown. In column C the numbers of the teams assigned to the group in question are 

entered. For example, the number 14 is entered for team A1 because Turkey is number 14 on the Language 

worksheet, and so on. 

 

11.4 Match data update 

The 'Matches' worksheet is shown. The kick-off times must be updated in column E and the venue numbers in 

column G. 

Should it be necessary to adapt the match numbers to the UEFA schedule, the corresponding changes can be 

made in column B after the sheet protection has been removed. 

 

11.5 Updating the match numbers on the schedule 

The match numbers are now adjusted on the ‘EURO’ worksheet. It is advisable to arrange the six group 

matches within each group in chronological order. 

That's it! - All these changes are automatically applied to the rest of the worksheets. 

Appendix: Special Case 'Direct Comparisons' 

Under UEFA rules for qualifying for the round of 16, direct comparisons take precedence over goal difference 

and number of goals scored. If there are still tied teams left after a direct comparison, the direct comparison 

will be reapplied to those teams. 

This special case shows an example in which after the direct comparison of all four teams, another direct 

comparison of three teams and then a third direct comparison of two teams takes place. 

The match results look like this: 

Netherlands Ukraine 2 : 2 

Austria North Macedonia 1 : 1 

Netherlands Austria 1 : 1 

Ukraine North Macedonia 0 : 0 

North Macedonia Netherlands 1 : 1 

Ukraine Austria 0 : 0 

 

The result is the following group table: 

 Pts. GD Goals 

Netherlands 3 0 4 : 4 

Austria 3 0 2 : 2 

North Macedonia 3 0 2 : 2 

Ukraine 3 0 2 : 2 

 

A direct comparison of all four teams shows: the Netherlands are in first place. 

Another direct comparison is made for the other three teams, which only takes into account the matches of 

these three against each other. The following table is used for this direct comparison: 

 Pts. GD Goals 

Austria 2 0 1 : 1 

North Macedonia 2 0 1 : 1 

Ukraine 2 0 0 : 0 

 

This puts Ukraine in last place. 

Another direct comparison is made for Austria and North Macedonia. 

Austria North Macedonia 1 : 1 

 

However, this does not lead to a decision, since the direct encounter between these two ended in a draw. 

Since the number of wins from all group matches is also the same, the fair play rating decides next. In the case 

of the same fair play rating, the rank in the European Qualifiers rankings then decides. Austria is in 16th place 

ahead of North Macedonia in 30th place (EURO 2020/2021). 


